
















































































































































































































































































































PLEASE NOTE:

I did not scan this ASSIST09 listing from the Motorola book. The listing was very 
large, and I was scanning a bound book which required each page to be scanned 
individually. Instead, I assembled the ASSIST09 source code using my own 
ASM09 assembler and placed the resultant listing into these pages. This has the 
added advantage that the text is searchable and can be extracted if you wish to use 
code snippets in other places. It does mean however that this listing is not precicely 
identical to the one originally printed in the Motorola MC6809-MC6809E  
Programming Reference manual.

Also note: I found the source code on which this listing is based via an internet 
search. It appears to be the original Motorola source code, however it had been 
modified both in function (code changes) and source format (different assembler). I 
have attempted to restore it as closely as possible to the original - please notify me if
you find errors. For the most part, the source was compatible with my assembler. I 
did have to change the single-quote character constants to double-quote format ( 'a' 
instead of 'a ), and some of the directives are slightly different. (TITLE instead of 
TTL for example).

Dave Dunfield

0000                         1  *************************************
0000                         2  * COPYRIGHT (C) MOTOROLA, INC. 1979 *
0000                         3  *************************************
0000                         4  
0000                         5  *************************************
0000                         6  *  THIS IS THE BASE ASSIST09 ROM.
0000                         7  *  IT MAY RUN WITH OR WITHOUT THE
0000                         8  *  EXTENSION ROM WHICH
0000                         9  *  WHEN PRESENT WILL BE AUTOMATICALLY
0000                        10  *  INCORPORATED BY THE BLDVTR
0000                        11  *  SUBROUTINE.
0000                        12  *************************************
0000                        13  
0000                        14  *********************************************
0000                        15  *         GLOBAL MODULE EQUATES
0000                        16  ********************************************
F800                        17  ROMBEG  EQU     $F800           ROM START ASSEMBLY ADDRESS
E700                        18  RAMOFS  EQU     -$1900          ROM OFFSET TO RAM WORK PAGE
0800                        19  ROMSIZ  EQU     2048            ROM SIZE
F000                        20  ROM2OF  EQU     ROMBEG-ROMSIZ   START OF EXTENSION ROM
E008                        21  ACIA    EQU     $E008           DEFAULT ACIA ADDRESS
E000                        22  PTM     EQU     $E000           DEFAULT PTM ADDRESS
0000                        23  DFTCHP  EQU     0               DEFAULT CHARACTER PAD COUNT
0005                        24  DFTNLP  EQU     5               DEFAULT NEW LINE PAD COUNT
003E                        25  PROMPT  EQU     '>'             PROMPT CHARACTER
0008                        26  NUMBKP  EQU     8               NUMBER OF BREAKPOINTS
0000                        27  *********************************************
0000                        28  
0000                        29  *********************************************
0000                        30  *  MISCELANEOUS EQUATES
0000                        31  *********************************************
0004                        32  EOT     EQU     $04             END OF TRANSMISSION
0007                        33  BELL    EQU     $07             BELL CHARACTER
000A                        34  LF      EQU     $0A             LINE FEED
000D                        35  CR      EQU     $0D             CARRIAGE RETURN
0010                        36  DLE     EQU     $10             DATA LINK ESCAPE
0018                        37  CAN     EQU     $18             CANCEL (CTL-X)
0000                        38  * PTM ACCESS DEFINITIONS
E001                        39  PTMSTA  EQU     PTM+1           READ STATUS REGISTER
E000                        40  PTMC13  EQU     PTM             CONTROL REGISTERS 1 AND 3
E001                        41  PTMC2   EQU     PTM+1           CONTROL REGISTER 2
E002                        42  PTMTM1  EQU     PTM+2           LATCH 1
E004                        43  PTMTM2  EQU     PTM+4           LATCH 2
E006                        44  PTMTM3  EQU     PTM+6           LATCH 3
0000                        45  
008C                        46  SKIP2   EQU     $8C             "CMPX #" OPCODE - SKIPS TWO BYTES
0000                        47  
0000                        48  *******************************************
0000                        49  *    ASSIST09 MONITOR SWI FUNCTIONS
0000                        50  * THE FOLLOWING EQUATES DEFINE FUNCTIONS PROVIDED
0000                        51  * BY THE ASSIST09 MONITOR VIA THE SWI INSTRUCTION.
0000                        52  ******************************************
0000                        53  INCHNP  EQU     0               INPUT CHAR IN A REG - NO PARITY
0001                        54  OUTCH   EQU     1               OUTPUT CHAR FROM A REG
0002                        55  PDATA1  EQU     2               OUTPUT STRING
0003                        56  PDATA   EQU     3               OUTPUT CR/LF THEN STRING
0004                        57  OUT2HS  EQU     4               OUTPUT TWO HEX AND SPACE
0005                        58  OUT4HS  EQU     5               OUTPUT FOUR HEX AND SPACE
0006                        59  PCRLF   EQU     6               OUTPUT CR/LF
0007                        60  SPACE   EQU     7               OUTPUT A SPACE
0008                        61  MONITR  EQU     8               ENTER ASSIST09 MONITOR
0009                        62  VCTRSW  EQU     9               VECTOR EXAMINE/SWITCH
000A                        63  BRKPT   EQU     10              USER PROGRAM BREAKPOINT
000B                        64  PAUSE   EQU     11              TASK PAUSE FUNCTION
000B                        65  NUMFUN  EQU     11              NUMBER OF AVAILABLE FUNCTIONS
0000                        66  * NEXT SUB-CODES FOR ACCESSING THE VECTOR TABLE.
0000                        67  * THEY ARE EQUIVALENT TO OFFSETS IN THE TABLE.
0000                        68  * RELATIVE POSITIONING MUST BE MAINTAINED.



DUNFIELD 6809 ASSEMBLER: ASSIST09                                     PAGE: 2

0000                        69  .AVTBL  EQU     0               ADDRESS OF VECTOR TABLE
0002                        70  .CMDL1  EQU     2               FIRST COMMAND LIST
0004                        71  .RSVD   EQU     4               RESERVED HARDWARE VECTOR
0006                        72  .SWI3   EQU     6               SWI3 ROUTINE
0008                        73  .SWI2   EQU     8               SWI2 ROUTINE
000A                        74  .FIRQ   EQU     10              FIRQ ROUTINE
000C                        75  .IRQ    EQU     12              IRQ ROUTINE
000E                        76  .SWI    EQU     14              SWI ROUTINE
0010                        77  .NMI    EQU     16              NMI ROUTINE
0012                        78  .RESET  EQU     18              RESET ROUTINE
0014                        79  .CION   EQU     20              CONSOLE ON
0016                        80  .CIDTA  EQU     22              CONSOLE INPUT DATA
0018                        81  .CIOFF  EQU     24              CONSOLE INPUT OFF
001A                        82  .COON   EQU     26              CONSOLE OUTPUT ON
001C                        83  .CODTA  EQU     28              CONSOLE OUTPUT DATA
001E                        84  .COOFF  EQU     30              CONSOLE OUTPUT OFF
0020                        85  .HSDTA  EQU     32              HIGH SPEED PRINTDATA
0022                        86  .BSON   EQU     34              PUNCH/LOAD ON
0024                        87  .BSDTA  EQU     36              PUNCH/LOAD DATA
0026                        88  .BSOFF  EQU     38              PUNCH/LOAD OFF
0028                        89  .PAUSE  EQU     40              TASK PAUSE ROUTINE
002A                        90  .EXPAN  EQU     42              EXPRESSION ANALYZER
002C                        91  .CMDL2  EQU     44              SECOND COMMAND LIST
002E                        92  .ACIA   EQU     46              ACIA ADDRESS
0030                        93  .PAD    EQU     48              CHARACTER PAD AND NEW LINE PAD
0032                        94  .ECHO   EQU     50              ECHO/LOAD AND NULL BKPT FLAG
0034                        95  .PTM    EQU     52              PTM ADDRESS
001B                        96  NUMVTR  EQU     52/2+1          NUMBER OF VECTORS
0034                        97  HIVTR   EQU     52              HIGHEST VECTOR OFFSET

0000                        99  ******************************************
0000                       100  *              WORK AREA
0000                       101  * THIS WORK AREA IS ASSIGNED TO THE PAGE ADDRESSED BY
0000                       102  * -$1800,PCR FROM THE BASE ADDRESS OF THE ASSIST09
0000                       103  * ROM.  THE DIRECT PAGE REGISTER DURING MOST ROUTINE
0000                       104  * OPERATIONS WILL POINT TO THIS WORK AREA.  THE STACK
0000                       105  * INITIALLY STARTS UNDER THE RESERVED WORK AREAS AS
0000                       106  * DEFINED HEREIN.
0000                       107  ******************************************
DF00                       108  WORKPG  EQU     ROMBEG+RAMOFS   SETUP DIRECT PAGE ADDRESS
0000                       109          SETDP   =WORKPG         NOTIFY ASSEMBLER
E000                       110          ORG     WORKPG+256      READY PAGE DEFINITIONS
E000                       111  * THE FOLLOWING THRU BKPTOP MUST RESIDE IN THIS ORDER
E000                       112  * FOR PROPER INITIALIZATION
DFFC                       113          ORG     *-4
DFFC                       114  PAUSER  EQU     *               PAUSE ROUTINE
DFFB                       115          ORG     *-1
DFFB                       116  SWIBFL  EQU     *               BYPASS SWI AS BREAKPOINT FLAG
DFFA                       117          ORG     *-1
DFFA                       118  BKPTCT  EQU     *               BREAKPOINT COUNT
DFF8                       119          ORG     *-2
DFF8                       120  SLEVEL  EQU     *               STACK TRACE LEVEL
DFC2                       121          ORG     *-(NUMVTR*2)
DFC2                       122  VECTAB  EQU     *               VECTOR TABLE
DFB2                       123          ORG     *-(2*NUMBKP)
DFB2                       124  BKPTBL  EQU     *               BREAKPOINT TABLE
DFA2                       125          ORG     *-(2*NUMBKP)
DFA2                       126  BKPTOP  EQU     *               BREAKPOINT OPCODE TABLE
DFA0                       127          ORG     *-2
DFA0                       128  WINDOW  EQU     *               WINDOW
DF9E                       129          ORG     *-2
DF9E                       130  ADDR    EQU     *               ADDRESS POINTER VALUE
DF9D                       131          ORG     *-1
DF9D                       132  BASEPG  EQU     *               BASE PAGE VALUE
DF9B                       133          ORG     *-2
DF9B                       134  NUMBER  EQU     *               BINARY BUILD AREA
DF99                       135          ORG     *-2
DF99                       136  LASTOP  EQU     *               LAST OPCODE TRACED
DF97                       137          ORG     *-2
DF97                       138  RSTACK  EQU     *               RESET STACK POINTER
DF95                       139          ORG     *-2
DF95                       140  PSTACK  EQU     *               COMMAND RECOVERY STACK
DF93                       141          ORG     *-2
DF93                       142  PCNTER  EQU     *               LAST PROGRAM COUNTER
DF91                       143          ORG     *-2
DF91                       144  TRACEC  EQU     *               TRACE COUNT
DF90                       145          ORG     *-1
DF90                       146  SWICNT  EQU     *               TRACE "SWI" NEST LEVEL COUNT
DF8F                       147          ORG     *-1             (MISFLG MUST FOLLOW SWICNT)
DF8F                       148  MISFLG  EQU     *               LOAD CMD/THRU BREAKPOINT FLAG
DF8E                       149          ORG     *-1
DF8E                       150  DELIM   EQU     *               EXPRESSION DELIMITER/WORK BYTE
DF66                       151          ORG     *-40
DF66                       152  ROM2WK  EQU     *               EXTENSION ROM RESERVED AREA
DF51                       153          ORG     *-21
DF51                       154  TSTACK  EQU     *               TEMPORARY STACK HOLD
DF51                       155  STACK   EQU     *               START OF INITIAL STACK

DF51                       157  ******************************************
DF51                       158  * DEFAULT THE ROM BEGINNING ADDRESS TO 'ROMBEG'
DF51                       159  * ASSIST09 IS POSITION ADDRESS INDEPENDENT, HOWEVER
DF51                       160  * WE ASSEMBLE ASSUMING CONTROL OF THE HARDWARE VECTORS.
DF51                       161  * NOTE THAT THE WORK RAM PAGE MUST BE 'RAMOFS'
DF51                       162  * FROM THE ROM BEGINNING ADDRESS.
DF51                       163  ********************************************
F800                       164          ORG     ROMBEG          ROM ASSEMBLY/DEFAULT ADDRESS
F800                       165  
F800                       166  *****************************************************
F800                       167  *            BLDVTR - BUILD ASSIST09 VECTOR TABLE
F800                       168  *  HARDWARE RESET CALLS THIS SUBROUTINE TO BUILD THE
F800                       169  *  ASSIST09 VECTOR TABLE.  THIS SUBROUTINE RESIDES AT
F800                       170  *  THE FIRST BYTE OF THE ASSIST09 ROM, AND CAN BE
F800                       171  *  CALLED VIA EXTERNAL CONTROL CODE FOR REMOTE
F800                       172  *  ASSIST09 EXECUTION.
F800                       173  * INPUT: S->VALID STACK RAM
F800                       174  * OUTPUT: U->VECTOR TABLE ADDRESS
F800                       175  *         DPR->ASSIST09 WORK AREA PAGE
F800                       176  *         THE VECTOR TABLE AND DEFAULTS ARE INITIALIZED
F800                       177  *  ALL REGISTERS VOLATILE
F800                       178  *************************************************
F800                       179  
F800  30 8D E7 BE          180  BLDVTR  LEAX    VECTAB,PCR      ADDRESS VECTOR TABLE
F804  1F 10                181          TFR     X,D             OBTAIN BASE PAGE ADDRESS
F806  1F 8B                182          TFR     A,DP            SETUP DPR
F808  97 9D                183          STA     BASEPG          STORE FOR QUICK REFERENCE
F80A  33 84                184          LEAU    ,X              RETURN TABLE TO CALLER
F80C  31 8C 35             185          LEAY    <INITVT,PCR     LOAD FROM ADDR
F80F  EF 81                186          STU     ,X++            INIT VECTOR TABLE ADDRESS
F811  C6 16                187          LDB     #NUMVTR-5       NUMBER RELOCATABLE VECTORS
F813  34 04                188          PSHS    B               STORE INDEX ON STACK
F815  1F 20                189  BLD2    TFR     Y,D             PREPARE ADDRESS RESOLVE
F817  E3 A1                190          ADDD    ,Y++            TO ABSOLUTE ADDRESS
F819  ED 81                191          STD     ,X++            INTO VECTOR TABLE
F81B  6A E4                192          DEC     ,S              COUNT DOWN
F81D  26 F6                193          BNE     BLD2            BRANCH IF MORE TO INSERT
F81F  C6 0D                194          LDB     #INTVE-INTVS    STATIC VALUE INIT LENGTH
F821  A6 A0                195  BLD3    LDA     ,Y+             LOAD NEXT BYTE
F823  A7 80                196          STA     ,X+             STORE INTO POSITION
F825  5A                   197          DECB                    COUNT DOWN
F826  26 F9                198          BNE     BLD3            LOOP UNTIL DONE
F828  31 8D F7 D4          199          LEAY    ROM2OF,PCR      TEST POSSIBLE EXTENSION ROM
F82C  8E 20 FE             200          LDX     #$20FE          LOAD "BRA *" FLAG PATTERN
F82F  AC A1                201          CMPX    ,Y++            ? EXTENDED ROM HERE
F831  26 02                202          BNE     BLDRTN          BRANCH NOT OUR ROM TO RETURN
F833  AD A4                203          JSR     ,Y              CALL EXTENDED ROM INITIALIZE
F835  35 84                204  BLDRTN  PULS    PC,B            RETURN TO INITIALIZER
F837                       205  
F837                       206  *****************************************************
F837                       207  *                 RESET ENTRY POINT
F837                       208  *  HARDWARE RESET ENTERS HERE IF ASSIST09 IS ENABLED
F837                       209  *  TO RECEIVE THE MC6809 HARDWARE VECTORS.  WE CALL
F837                       210  *  THE BLDVTR SUBROUTINE TO INITIALIZE THE VECTOR
F837                       211  *  TABLE, STACK, AND THEN FIREUP THE MONITOR VIA SWI
F837                       212  *  CALL.
F837                       213  *******************************************************
F837  32 8D E7 16          214  RESET   LEAS    STACK,PCR       SETUP INITIAL STACK
F83B  8D C3                215          BSR     BLDVTR          BUILD VECTOR TABLE
F83D  4F                   216  RESET2  CLRA                    ISSUE STARTUP MESSAGE
F83E  1F 8B                217          TFR     A,DP            DEFAULT TO PAGE ZERO
F840  3F                   218          SWI                     PERFORM MONITOR FIREUP
F841  08                   219          FCB     MONITR          TO ENTER COMMAND PROCESSING
F842  20 F9                220          BRA     RESET2          REENTER MONITOR IF 'CONTINUE'
F844                       221  
F844                       222  ******************************************************
F844                       223  *         INITVT - INITIAL VECTOR TABLE
F844                       224  *  THIS TABLE IS RELOCATED TO RAM AND REPRESENTS THE

F844                       225  *  INITIAL STATE OF THE VECTOR TABLE. ALL ADDRESSES
F844                       226  *  ARE CONVERTED TO ABSOLUTE FORM.  THIS TABLE STARTS
F844                       227  *  WITH THE SECOND ENTRY, ENDS WITH STATIC CONSTANT
F844                       228  *  INITIALIZATION DATA WHICH CARRIES BEYOND THE TABLE.
F844                       229  ************************************************
F844  01 58                230  INITVT  FDB     CMDTBL-*        DEFAULT FIRST COMMAND TABLE
F846  02 92                231          FDB     RSRVDR-*        DEFAULT UNDEFINED HARDWARE VECTOR
F848  02 90                232          FDB     SWI3R-*         DEFAULT SWI3
F84A  02 8E                233          FDB     SWI2R-*         DEFAULT SWI2
F84C  02 70                234          FDB     FIRQR-*         DEFAULT FIRQ
F84E  02 8A                235          FDB     IRQR-*          DEFAULT IRQ ROUTINE
F850  00 45                236          FDB     SWIR-*          DEFAULT SWI ROUTINE
F852  02 2B                237          FDB     NMIR-*          DEFAULT NMI ROUTINE
F854  FF E3                238          FDB     RESET-*         RESTART VECTOR
F856  02 90                239          FDB     CION-*          DEFAULT CION
F858  02 84                240          FDB     CIDTA-*         DEFAULT CIDTA
F85A  02 96                241          FDB     CIOFF-*         DEFAULT CIOFF
F85C  02 8A                242          FDB     COON-*          DEFAULT COON
F85E  02 93                243          FDB     CODTA-*         DEFAULT CODTA
F860  02 90                244          FDB     COOFF-*         DEFAULT COOFF
F862  03 9A                245          FDB     HSDTA-*         DEFAULT HSDTA
F864  02 B7                246          FDB     BSON-*          DEFAULT BSON
F866  02 D2                247          FDB     BSDTA-*         DEFAULT BSDTA
F868  02 BF                248          FDB     BSOFF-*         DEFAULT BSOFF
F86A  E7 92                249          FDB     PAUSER-*        DEFAULT PAUSE ROUTINE
F86C  04 7D                250          FDB     EXP1-*          DEFAULT EXPRESSION ANALYZER
F86E  01 2D                251          FDB     CMDTB2-*        DEFAULT SECOND COMMAND TABLE
F870                       252  * CONSTANTS
F870  E0 08                253  INTVS   FDB     ACIA            DEFAULT ACIA
F872  00 05                254          FCB     DFTCHP,DFTNLP   DEFAULT NULL PADDS
F874  00 00                255          FDB     0               DEFAULT ECHO
F876  E0 00                256          FDB     PTM             DEFAULT PTM
F878  00 00                257          FDB     0               INITIAL STACK TRACE LEVEL
F87A  00                   258          FCB     0               INITIAL BREAKPOINT COUNT
F87B  00                   259          FCB     0               SWI BREAKPOINT LEVEL
F87C  39                   260          FCB     $39             DEFAULT PAUSE ROUTINE (RTS)
F87D                       261  INTVE   EQU     *
F87D                       262  *B
F87D                       263  
F87D                       264  ***********************************************
F87D                       265  *             ASSIST09 SWI HANDLER
F87D                       266  *  THE SWI HANDLER PROVIDES ALL INTERFACING NECESSARY
F87D                       267  *  FOR A USER PROGRAM.  A FUNCTION BYTE IS ASSUMED TO
F87D                       268  *  FOLLOW THE SWI INSTRUCTION.  IT IS BOUND CHECKED
F87D                       269  *  AND THE PROPER ROUTINE IS GIVEN CONTROL.  THIS
F87D                       270  *  INVOCATION MAY ALSO BE A BREAKPOINT INTERRUPT.
F87D                       271  *  IF SO, THE BREAKPOINT HANDLER IS ENTERED.
F87D                       272  * INPUT: MACHINE STATE DEFINED FOR SWI
F87D                       273  * OUTPUT: VARIES ACCORDING TO FUNCTION CALLED. PC ON
F87D                       274  *     CALLERS STACK INCREMENTED BY ONE IF VALID CALL.
F87D                       275  * VOLATILE REGISTERS: SEE FUNCTIONS CALLED
F87D                       276  * STATE: RUNS DISABLED UNLESS FUNCTION CLEARS I FLAG.
F87D                       277  ************************************************
F87D                       278  
F87D                       279  * SWI FUNCTION VECTOR TABLE
F87D  01 94                280  SWIVTB  FDB     ZINCH-SWIVTB    INCHNP
F87F  01 B1                281          FDB     ZOTCH1-SWIVTB   OUTCH
F881  01 CB                282          FDB     ZPDTA1-SWIVTB   PDATA1
F883  01 C3                283          FDB     ZPDATA-SWIVTB   PDATA
F885  01 75                284          FDB     ZOT2HS-SWIVTB   OUT2HS
F887  01 73                285          FDB     ZOT4HS-SWIVTB   OUT4HS
F889  01 C0                286          FDB     ZPCRLF-SWIVTB   PCRLF
F88B  01 79                287          FDB     ZSPACE-SWIVTB   SPACE
F88D  00 55                288          FDB     ZMONTR-SWIVTB   MONITR
F88F  01 7D                289          FDB     ZVSWTH-SWIVTB   VCTRSW
F891  02 56                290          FDB     ZBKPNT-SWIVTB   BREAKPOINT
F893  01 D1                291          FDB     ZPAUSE-SWIVTB   TASK PAUSE
F895                       292  

F895  6A 8D E6 F7          293  SWIR    DEC     SWICNT,PCR      UP "SWI" LEVEL FOR TRACE
F899  17 02 25             294          LBSR    LDDP            SETUP PAGE AND VERIFY STACK
F89C                       295  * CHECK FOR BREAKPOINT TRAP
F89C  EE 6A                296          LDU     10,S            LOAD PROGRAM COUNTER
F89E  33 5F                297          LEAU    -1,U            BACK TO SWI ADDRESS
F8A0  0D FB                298          TST     SWIBFL          ? THIS "SWI" BREAKPOINT
F8A2  26 11                299          BNE     SWIDNE          BRANCH IF SO TO LET THROUGH
F8A4  17 06 9B             300          LBSR    CBKLDR          OBTAIN BREAKPOINT POINTERS
F8A7  50                   301          NEGB                    OBTAIN POSITIVE COUNT
F8A8  5A                   302  SWILP   DECB                    COUNT DOWN
F8A9  2B 0A                303          BMI     SWIDNE          BRANCH WHEN DONE
F8AB  11 A3 A1             304          CMPU    ,Y++            ? WAS THIS A BREAKPOINT
F8AE  26 F8                305          BNE     SWILP           BRANCH IF NOT
F8B0  EF 6A                306          STU     10,S            SET PROGRAM COUNTER BACK
F8B2  16 02 1E             307          LBRA    ZBKPNT          GO DO BREAKPOINT
F8B5  0F FB                308  SWIDNE  CLR     SWIBFL          CLEAR IN CASE SET
F8B7  37 06                309          PULU    D               OBTAIN FUNCTION BYTE, UP PC
F8B9  C1 0B                310          CMPB    #NUMFUN         ? TOO HIGH
F8BB  10 22 02 0F          311          LBHI    ERROR           YES, DO BREAKPOINT
F8BF  EF 6A                312          STU     10,S            BUMP PROGRAM COUNTER PAST SWI
F8C1  58                   313          ASLB    FUNCTION        CODE TIMES TWO
F8C2  33 8C B8             314          LEAU    SWIVTB,PCR      OBTAIN VECTOR BRANCH ADDRESS
F8C5  EC C5                315          LDD     B,U             LOAD OFFSET
F8C7  6E CB                316          JMP     D,U             JUMP TO ROUTINE
F8C9                       317  
F8C9                       318  **********************************************
F8C9                       319  * REGISTERS TO FUNCTION ROUTINES:
F8C9                       320  *  DP-> WORK AREA PAGE
F8C9                       321  *  D,Y,U=UNRELIABLE           X=AS CALLED FROM USER
F8C9                       322  *  S=AS FROM SWI INTERRUPT
F8C9                       323  *********************************************
F8C9                       324  
F8C9                       325  **************************************************
F8C9                       326  *            [SWI FUNCTION 8]
F8C9                       327  *              MONITOR ENTRY
F8C9                       328  *  FIREUP THE ASSIST09 MONITOR.
F8C9                       329  *  THE STACK WITH ITS VALUES FOR THE DIRECT PAGE
F8C9                       330  *  REGISTER AND CONDITION CODE FLAGS ARE USED AS IS.
F8C9                       331  *   1) INITIALIZE CONSOLE I/O
F8C9                       332  *   2) OPTIONALLY PRINT SIGNON
F8C9                       333  *   3) INITIALIZE PTM FOR SINGLE STEPPING
F8C9                       334  *   4) ENTER COMMAND PROCESSOR
F8C9                       335  * INPUT: A=0 INIT CONSOLE AND PRINT STARTUP MESSAGE
F8C9                       336  *        A#0 OMIT CONSOLE INIT AND STARTUP MESSAGE
F8C9                       337  *************************************************
F8C9                       338  
F8C9  41 53 53 49 53 54 +  339  SIGNON  FCC     /ASSIST09/      SIGNON EYE-CATCHER
F8D1  04                   340          FCB     EOT
F8D2                       341  
F8D2  10 DF 97             342  ZMONTR  STS     RSTACK          SAVE FOR BAD STACK RECOVERY
F8D5  6D 61                343          TST     1,S             ? INIT CONSOLE AND SEND MSG
F8D7  26 0D                344          BNE     ZMONT2          BRANCH IF NOT
F8D9  AD 9D E6 F9          345          JSR     [VECTAB+.CION,PCR]      READY CONSOLE INPUT
F8DD  AD 9D E6 FB          346          JSR     [VECTAB+.COON,PCR]      READY CONSOLE OUTPUT
F8E1  30 8C E5             347          LEAX    SIGNON,PCR      READY SIGNON EYE-CATCHER
F8E4  3F                   348          SWI                     PERFORM
F8E5  03                   349          FCB     PDATA           PRINT STRING
F8E6  9E F6                350  ZMONT2  LDX     VECTAB+.PTM     LOAD PTM ADDRESS
F8E8  27 0D                351          BEQ     CMD             BRANCH IF NOT TO USE A PTM
F8EA  6F 02                352          CLR     PTMTM1-PTM,X    SET LATCH TO CLEAR RESET
F8EC  6F 03                353          CLR     PTMTM1+1-PTM,X  AND SET GATE HIGH
F8EE  CC 01 A6             354          LDD     #$01A6          SETUP TIMER 1 MODE
F8F1  A7 01                355          STA     PTMC2-PTM,X     SETUP FOR CONTROL REGISTER1
F8F3  E7 00                356          STB     PTMC13-PTM,X    SET OUTPUT ENABLED/
F8F5                       357  *    SINGLE SHOT/ DUAL 8 BIT/INTERNAL MODE/OPERATE
F8F5  6F 01                358          CLR     PTMC2-PTM,X     SET CR2 BACK TO RESET FORM
F8F7                       359  * FALL INTO COMMAND PROCESSOR
F8F7                       360  

F8F7                       361  ***************************************************
F8F7                       362  *          COMMAND HANDLER
F8F7                       363  *  BREAKPOINTS ARE REMOVED AT THIS TIME.
F8F7                       364  *  PROMPT FOR A COMMAND, AND STORE ALL CHARACTERS
F8F7                       365  *  UNTIL A SEPARATOR ON THE STACK.
F8F7                       366  *  SEARCH FOR FIRST MATCHING COMMAND SUBSET,
F8F7                       367  *  CALL IT OR GIVE '?' RESPONSE.
F8F7                       368  *  DURING COMMAND SEARCH:
F8F7                       369  *      B=OFFSET TO NEXT ENTRY ON X
F8F7                       370  *      U=SAVED S
F8F7                       371  *      U-1=ENTRY SIZE+2
F8F7                       372  *      U-2=VALID NUMBER FLAG (>=0 VALID)/COMPARE CNT
F8F7                       373  *      U-3=CARRIAGE RETURN FLAG (0=CR HAS BEEN DONE)
F8F7                       374  *      U-4=START OF COMMAND STORE
F8F7                       375  *      S+0=END OF COMMAND STORE
F8F7                       376  ***********************************************
F8F7  3F                   377  CMD     SWI                     TO NEW LINE
F8F8  06                   378          FCB     PCRLF           FUNCTION
F8F9                       379  * DISARM THE BREAKPOINTS
F8F9  17 06 46             380  CMDNEP  LBSR    CBKLDR          OBTAIN BREAKPOINT POINTERS
F8FC  2A 0C                381          BPL     CMDNOL          BRANCH IF NOT ARMED OR NONE
F8FE  50                   382          NEGB                    MAKE POSITIVE
F8FF  D7 FA                383          STB     BKPTCT          FLAG AS DISARMED
F901  5A                   384  CMDDDL  DECB                    ? FINISHED
F902  2B 06                385          BMI     CMDNOL          BRANCH IF SO
F904  A6 30                386          LDA     -NUMBKP*2,Y     LOAD OPCODE STORED
F906  A7 B1                387          STA     [,Y++]          STORE BACK OVER "SWI"
F908  20 F7                388          BRA     CMDDDL          LOOP UNTIL DONE
F90A  AE 6A                389  CMDNOL  LDX     10,S            LOAD USERS PROGRAM COUNTER
F90C  9F 93                390          STX     PCNTER          SAVE FOR EXPRESSION ANALYZER
F90E  86 3E                391          LDA     #PROMPT         LOAD PROMPT CHARACTER
F910  3F                   392          SWI                     SEND TO OUTPUT HANDLER
F911  01                   393          FCB     OUTCH           FUNCTION
F912  33 E4                394          LEAU    ,S              REMEMBER STACK RESTORE ADDRESS
F914  DF 95                395          STU     PSTACK          REMEMBER STACK FOR ERROR USE
F916  4F                   396          CLRA                    PREPARE ZERO
F917  5F                   397          CLRB                    PREPARE ZERO
F918  DD 9B                398          STD     NUMBER          CLEAR NUMBER BUILD AREA
F91A  DD 8F                399          STD     MISFLG          CLEAR MISCEL. AND SWICNT FLAGS
F91C  DD 91                400          STD     TRACEC          CLEAR TRACE COUNT
F91E  C6 02                401          LDB     #2              SET D TO TWO
F920  34 07                402          PSHS    D,CC            PLACE DEFAULTS ONTO STACK
F922                       403  * CHECK FOR "QUICK" COMMANDS.
F922  17 04 54             404          LBSR    READ            OBTAIN FIRST CHARACTER
F925  30 8D 05 81          405          LEAX    CDOT+2,PCR      PRESET FOR SINGLE TRACE
F929  81 2E                406          CMPA    #'.'            ? QUICK TRACE
F92B  27 5A                407          BEQ     CMDXQT          BRANCH EQUAL FOR TRACE ONE
F92D  30 8D 04 E9          408          LEAX    CMPADP+2,PCR    READY MEMORY ENTRY POINT
F931  81 2F                409          CMPA    #'/'            ? OPEN LAST USED MEMORY
F933  27 52                410          BEQ     CMDXQT          BRANCH TO DO IT IF SO
F935                       411  * PROCESS NEXT CHARACTER
F935  81 20                412  CMD2    CMPA    #' '            ? BLANK OR DELIMITER
F937  23 14                413          BLS     CMDGOT          BRANCH YES, WE HAVE IT
F939  34 02                414          PSHS    A               BUILD ONTO STACK
F93B  6C 5F                415          INC     -1,U            COUNT THIS CHARACTER
F93D  81 2F                416          CMPA    #'/'            ? MEMORY COMMAND
F93F  27 4F                417          BEQ     CMDMEM          BRANCH IF SO
F941  17 04 0B             418          LBSR    BLDHXC          TREAT AS HEX VALUE
F944  27 02                419          BEQ     CMD3            BRANCH IF STILL VALID NUMBER
F946  6A 5E                420          DEC     -2,U            FLAG AS INVALID NUMBER
F948  17 04 2E             421  CMD3    LBSR    READ            OBTAIN NEXT CHARACTER
F94B  20 E8                422          BRA     CMD2            TEST NEXT CHARACTER
F94D                       423  * GOT COMMAND, NOW SEARCH TABLES
F94D  80 0D                424  CMDGOT  SUBA    #CR             SET ZERO IF CARRIAGE RETURN
F94F  A7 5D                425          STA     -3,U            SETUP FLAG
F951  9E C4                426          LDX     VECTAB+.CMDL1   START WITH FIRST CMD LIST
F953  E6 80                427  CMDSCH  LDB     ,X+             LOAD ENTRY LENGTH
F955  2A 10                428          BPL     CMDSME          BRANCH IF NOT LIST END

F957  9E EE                429          LDX     VECTAB+.CMDL2   NOW TO SECOND CMD LIST
F959  5C                   430          INCB                    ? TO CONTINUE TO DEFAULT LIST
F95A  27 F7                431          BEQ     CMDSCH          BRANCH IF SO
F95C  10 DE 95             432  CMDBAD  LDS     PSTACK          RESTORE STACK
F95F  30 8D 01 5A          433          LEAX    ERRMSG,PCR      POINT TO ERROR STRING
F963  3F                   434          SWI                     SEND OUT
F964  02                   435          FCB     PDATA1          TO CONSOLE
F965  20 90                436          BRA     CMD             AND TRY AGAIN
F967                       437  * SEARCH NEXT ENTRY
F967  5A                   438  CMDSME  DECB                    TAKE ACCOUNT OF LENGTH BYTE
F968  E1 5F                439          CMPB    -1,U            ? ENTERED LONGER THAN ENTRY
F96A  24 03                440          BHS     CMDSIZ          BRANCH IF NOT TOO LONG
F96C  3A                   441  CMDFLS  ABX     SKIP            TO NEXT ENTRY
F96D  20 E4                442          BRA     CMDSCH          AND TRY NEXT
F96F  31 5D                443  CMDSIZ  LEAY    -3,U            PREPARE TO COMPARE
F971  A6 5F                444          LDA     -1,U            LOAD SIZE+2
F973  80 02                445          SUBA    #2              TO ACTUAL SIZE ENTERED
F975  A7 5E                446          STA     -2,U            SAVE SIZE FOR COUNTDOWN
F977  5A                   447  CMDCMP  DECB                    DOWN ONE BYTE
F978  A6 80                448          LDA     ,X+             NEXT COMMAND CHARACTER
F97A  A1 A2                449          CMPA    ,-Y             ? SAME AS THAT ENTERED
F97C  26 EE                450          BNE     CMDFLS          BRANCH TO FLUSH  IF NOT
F97E  6A 5E                451          DEC     -2,U            COUNT DOWN LENGTH OF ENTRY
F980  26 F5                452          BNE     CMDCMP          BRANCH IF MORE TO TEST
F982  3A                   453          ABX     TO              NEXT ENTRY
F983  EC 1E                454          LDD     -2,X            LOAD OFFSET
F985  30 8B                455          LEAX    D,X             COMPUTE ROUTINE ADDRESS+2
F987  6D 5D                456  CMDXQT  TST     -3,U            SET CC FOR CARRIAGE RETURN TEST
F989  32 C4                457          LEAS    ,U              DELETE STACK WORK AREA
F98B  AD 1E                458          JSR     -2,X            CALL COMMAND
F98D  16 FF 7A             459          LBRA    CMDNOL          GO GET NEXT COMMAND
F990  6D 5E                460  CMDMEM  TST     -2,U            ? VALID HEX NUMBER ENTERED
F992  2B C8                461          BMI     CMDBAD          BRANCH ERROR IF NOT
F994  30 88 AE             462          LEAX    <CMEMN-CMPADP,X TO DIFFERENT ENTRY
F997  DC 9B                463          LDD     NUMBER          LOAD NUMBER ENTERED
F999  20 EC                464          BRA     CMDXQT          AND ENTER MEMORY COMMAND
F99B                       465  
F99B                       466  ** COMMANDS ARE ENTERED AS A SUBROUTINE WITH:
F99B                       467  **    DPR->ASSIST09 DIRECT PAGE WORK AREA
F99B                       468  **    Z=1 CARRIAGE RETURN ENTERED
F99B                       469  **    Z=0 NON CARRIAGE RETURN DELIMITER
F99B                       470  **    S=NORMAL RETURN ADDRESS
F99B                       471  ** THE LABEL "CMDBAD" MAY BE ENTERED TO ISSUE AN
F99B                       472  ** AN ERROR FLAG (*).
F99B                       473  
F99B                       474  **************************************************
F99B                       475  *        ASSIST09 COMMAND TABLES
F99B                       476  *  THESE ARE THE DEFAULT COMMAND TABLES.  EXTERNAL
F99B                       477  *  TABLES OF THE SAME FORMAT MAY EXTEND/REPLACE
F99B                       478  *  THESE BY USING THE VECTOR SWAP FUNCTION.
F99B                       479  *
F99B                       480  * ENTRY FORMAT:
F99B                       481  *     +0...TOTAL SIZE OF ENTRY (INCLUDING THIS BYTE)
F99B                       482  *     +1...COMMAND STRING
F99B                       483  *     +N...TWO BYTE OFFSET TO COMMAND (ENTRYADDR-*)
F99B                       484  *
F99B                       485  *  THE TABLES TERMINATE WITH A ONE BYTE -1 OR -2.
F99B                       486  *  THE -1 CONTINUES THE COMMAND SEARCH WITH THE
F99B                       487  *         SECOND COMMAND TABLE.
F99B                       488  *  THE -2 TERMINATES COMMAND SEARCHES.
F99B                       489  *****************************************************
F99B                       490  
F99B                       491  * THIS IS THE DEFAULT LIST FOR THE SECOND COMMAND
F99B                       492  * LIST ENTRY.
F99B  FE                   493  CMDTB2  FCB     -2              STOP COMMAND SEARCHES
F99C                       494  
F99C                       495  * THIS IS THE DEFAULT LIST FOR THE FIRST COMMAND
F99C                       496  * LIST ENTRY.

F99C                       497  CMDTBL  EQU     *               MONITOR COMMAND TABLE
F99C  04                   498          FCB     4
F99D  42                   499          FCC     /B/             'BREAKPOINT' COMMAND
F99E  05 4D                500          FDB     CBKPT-*
F9A0  04                   501          FCB     4
F9A1  43                   502          FCC     /C/             'CALL' COMMAND
F9A2  04 17                503          FDB     CCALL-*
F9A4  04                   504          FCB     4
F9A5  44                   505          FCC     /D/             'DISPLAY' COMMAND
F9A6  04 9D                506          FDB     CDISP-*
F9A8  04                   507          FCB     4
F9A9  45                   508          FCC     /E/             'ENCODE' COMMAND
F9AA  05 9F                509          FDB     CENCDE-*
F9AC  04                   510          FCB     4
F9AD  47                   511          FCC     /G/             'GO' COMMAND
F9AE  03 D2                512          FDB     CGO-*
F9B0  04                   513          FCB     4
F9B1  4C                   514          FCC     /L/             'LOAD' COMMAND
F9B2  04 DD                515          FDB     CLOAD-*
F9B4  04                   516          FCB     4
F9B5  4D                   517          FCC     /M/             'MEMORY' COMMAND
F9B6  04 0D                518          FDB     CMEM-*
F9B8  04                   519          FCB     4
F9B9  4E                   520          FCC     /N/             'NULLS' COMMAND
F9BA  04 FD                521          FDB     CNULLS-*
F9BC  04                   522          FCB     4
F9BD  4F                   523          FCC     /O/             'OFFSET' COMMAND
F9BE  05 0A                524          FDB     COFFS-*
F9C0  04                   525          FCB     4
F9C1  50                   526          FCC     /P/             'PUNCH' COMMAND
F9C2  04 AF                527          FDB     CPUNCH-*
F9C4  04                   528          FCB     4
F9C5  52                   529          FCC     /R/             'REGISTERS' COMMAND
F9C6  02 84                530          FDB     CREG-*
F9C8  04                   531          FCB     4
F9C9  53                   532          FCC     /S/             'STLEVEL' COMMAND
F9CA  04 F2                533          FDB     CSTLEV-*
F9CC  04                   534          FCB     4
F9CD  54                   535          FCC     /T/             'TRACE' COMMAND
F9CE  04 D6                536          FDB     CTRACE-*
F9D0  04                   537          FCB     4
F9D1  56                   538          FCC     /V/             'VERIFY' COMMAND
F9D2  04 CF                539          FDB     CVER-*
F9D4  04                   540          FCB     4
F9D5  57                   541          FCC     /W/             'WINDOW' COMMAND
F9D6  04 68                542          FDB     CWINDO-*
F9D8  FF                   543          FCB     -1              END, CONTINUE WITH THE SECOND
F9D9                       544  
F9D9                       545  *************************************************
F9D9                       546  *             [SWI FUNCTIONS 4 AND 5]
F9D9                       547  *      4 - OUT2HS - DECODE BYTE TO HEX AND ADD SPACE
F9D9                       548  *      5 - OUT4HS - DECODE WORD TO HEX AND ADD SPACE
F9D9                       549  * INPUT: X->BYTE OR WORD TO DECODE
F9D9                       550  * OUTPUT: CHARACTERS SENT TO OUTPUT HANDLER
F9D9                       551  *         X->NEXT BYTE OR WORD
F9D9                       552  **************************************************
F9D9                       553  
F9D9  A6 80                554  ZOUT2H  LDA     ,X+             LOAD NEXT BYTE
F9DB  34 06                555          PSHS    D               SAVE - DO NOT REREAD
F9DD  C6 10                556          LDB     #16             SHIFT BY 4 BITS
F9DF  3D                   557          MUL                     WITH MULTIPLY
F9E0  8D 04                558          BSR     ZOUTHX          SEND OUT AS HEX
F9E2  35 06                559          PULS    D               RESTORE BYTES
F9E4  84 0F                560          ANDA    #$0F            ISOLATE RIGHT HEX
F9E6  8B 90                561  ZOUTHX  ADDA    #$90            PREPARE A-F ADJUST
F9E8  19                   562          DAA     ADJUST
F9E9  89 40                563          ADCA    #$40            PREPARE CHARACTER BITS
F9EB  19                   564          DAA     ADJUST

F9EC  6E 9D E5 EE          565  SEND    JMP     [VECTAB+.CODTA,PCR]     SEND TO OUT HANDLER
F9F0                       566  
F9F0  8D E7                567  ZOT4HS  BSR     ZOUT2H          CONVERT FIRST BYTE
F9F2  8D E5                568  ZOT2HS  BSR     ZOUT2H          CONVERT BYTE TO HEX
F9F4  AF 64                569          STX     4,S             UPDATE USERS X REGISTER
F9F6                       570  * FALL INTO SPACE ROUTINE
F9F6                       571  
F9F6                       572  *************************************************
F9F6                       573  *            [SWI FUNCTION 7]
F9F6                       574  *         SPACE - SEND BLANK TO OUTPUT HANDLER
F9F6                       575  * INPUT: NONE
F9F6                       576  * OUTPUT: BLANK SEND TO CONSOLE HANDLER
F9F6                       577  *************************************************
F9F6  86 20                578  ZSPACE  LDA     #' '            LOAD BLANK
F9F8  20 3D                579          BRA     ZOTCH2          SEND AND RETURN
F9FA                       580  
F9FA                       581  ***********************************************
F9FA                       582  *             [SWI FUNCTION 9]
F9FA                       583  *          SWAP VECTOR TABLE ENTRY
F9FA                       584  * INPUT: A=VECTOR TABLE CODE (OFFSET)
F9FA                       585  *        X=0 OR REPLACEMENT VALUE
F9FA                       586  * OUTPUT: X=PREVIOUS VALUE
F9FA                       587  ***********************************************
F9FA  A6 61                588  ZVSWTH  LDA     1,S             LOAD REQUESTERS A
F9FC  81 34                589          CMPA    #HIVTR          ? SUB-CODE TOO HIGH
F9FE  22 39                590          BHI     ZOTCH3          IGNORE CALL IF SO
FA00  10 9E C2             591          LDY     VECTAB+.AVTBL   LOAD VECTOR TABLE ADDRESS
FA03  EE A6                592          LDU     A,Y             U=OLD ENTRY
FA05  EF 64                593          STU     4,S             RETURN OLD VALUE TO CALLERS X
FA07  AF 7E                594          STX     -2,S            ? X=0
FA09  27 2E                595          BEQ     ZOTCH3          YES, DO NOT CHANGE ENTRY
FA0B  AF A6                596          STX     A,Y             REPLACE ENTRY
FA0D  20 2A                597          BRA     ZOTCH3          RETURN FROM SWI
FA0F                       598  *D
FA0F                       599  
FA0F                       600  ************************************************
FA0F                       601  *                    [SWI FUNCTION 0]
FA0F                       602  *   INCHNP - OBTAIN INPUT CHAR IN A (NO PARITY)
FA0F                       603  *  NULLS AND RUBOUTS ARE IGNORED.
FA0F                       604  *  AUTOMATIC LINE FEED IS SENT UPON RECIEVING A
FA0F                       605  *      CARRIAGE RETURN.
FA0F                       606  *  UNLESS WE ARE LOADING FROM TAPE.
FA0F                       607  ************************************************
FA0F  8D 5D                608  ZINCHP  BSR     XQPAUS          RELEASE PROCESSOR
FA11  8D 5F                609  ZINCH   BSR     XQCIDT          CALL INPUT DATA APPENDAGE
FA13  24 FA                610          BCC     ZINCHP          LOOP IF NONE AVAILABLE
FA15  4D                   611          TSTA    ?               TEST FOR NULL
FA16  27 F9                612          BEQ     ZINCH           IGNORE NULL
FA18  81 7F                613          CMPA    #$7F            ? RUBOUT
FA1A  27 F5                614          BEQ     ZINCH           BRANCH YES TO IGNORE
FA1C  A7 61                615          STA     1,S             STORE INTO CALLERS A
FA1E  0D 8F                616          TST     MISFLG          ? LOAD IN PROGRESS
FA20  26 17                617          BNE     ZOTCH3          BRANCH IF SO TO NOT ECHO
FA22  81 0D                618          CMPA    #CR             ? CARRIAGE RETURN
FA24  26 04                619          BNE     ZIN2            NO, TEST ECHO BYTE
FA26  86 0A                620          LDA     #LF             LOAD LINE FEED
FA28  8D C2                621          BSR     SEND            ALWAYS ECHO LINE FEED
FA2A  0D F4                622  ZIN2    TST     VECTAB+.ECHO    ? ECHO DESIRED
FA2C  26 0B                623          BNE     ZOTCH3          NO, RETURN
FA2E                       624  * FALL THROUGH TO OUTCH
FA2E                       625  
FA2E                       626  ************************************************
FA2E                       627  *              [SWI FUNCTION 1]
FA2E                       628  *          OUTCH - OUTPUT CHARACTER FROM A
FA2E                       629  *  INPUT:  NONE
FA2E                       630  *  OUTPUT: IF LINEFEED IS THE OUTPUT CHARACTER THEN
FA2E                       631  *           C=0 NO CTL-X RECIEVED, C=1 CTL-X RECIEVED
FA2E                       632  ************************************************

FA2E  A6 61                633  ZOTCH1  LDA     1,S             LOAD CHARACTER TO SEND
FA30  30 8C 09             634          LEAX    <ZPCRLS,PCR     DEFAULT FOR LINE FEED
FA33  81 0A                635          CMPA    #LF             ? LINE FEED
FA35  27 0F                636          BEQ     ZPDTLP          BRANCH TO CHECK PAUSE IF SO
FA37  8D B3                637  ZOTCH2  BSR     SEND            SEND TO OUTPUT ROUTINE
FA39  0C 90                638  ZOTCH3  INC     SWICNT          BUMP UP "SWI" TRACE NEST LEVEL
FA3B  3B                   639          RTI     RETURN          FROM "SWI" FUNCTION
FA3C                       640  
FA3C                       641  **************************************************
FA3C                       642  *              [SWI FUNCTION 6]
FA3C                       643  *        PCRLF - SEND CR/LF TO CONSOLE HANDLER
FA3C                       644  *  INPUT: NONE
FA3C                       645  *  OUTPUT: CR AND LF SENT TO HANDLER
FA3C                       646  *          C=0 NO CTL-X, C=1 CTL-X RECIEVED
FA3C                       647  **************************************************
FA3C                       648  
FA3C  04                   649  ZPCRLS  FCB     EOT             NULL STRING
FA3D                       650  
FA3D  30 8C FC             651  ZPCRLF  LEAX    ZPCRLS,PCR      READY CR,LF STRING
FA40                       652  * FALL INTO CR/LF CODE
FA40                       653  
FA40                       654  **************************************************
FA40                       655  *             [SWI FUNCTION 3]
FA40                       656  *        PDATA - OUTPUT CR/LF AND STRING
FA40                       657  * INPUT: X->STRING
FA40                       658  * OUTPUT: CR/LF AND STRING SENT TO OUTPUT CONSOLE
FA40                       659  *         HANDLER.
FA40                       660  *     C=0 NO CTL-X, C=1 CTL-X RECIEVED
FA40                       661  * NOTE: LINE FEED MUST FOLLOW CARRIAGE RETURN FOR
FA40                       662  *       PROPER PUNCH DATA.
FA40                       663  **************************************************
FA40  86 0D                664  ZPDATA  LDA     #CR             LOAD CARRIAGE RETURN
FA42  8D A8                665          BSR     SEND            SEND IT
FA44  86 0A                666          LDA     #LF             LOAD LINE FEED
FA46                       667  * FALL INTO PDATA1
FA46                       668  
FA46                       669  *************************************************
FA46                       670  *             [SWI FUNCTION 2]
FA46                       671  *         PDATA1 - OUTPUT STRING TILL EOT ($04)
FA46                       672  *  THIS ROUTINE PAUSES IF AN INPUT BYTE BECOMES
FA46                       673  *  AVAILABLE DURING OUTPUT TRANSMISSION UNTIL A
FA46                       674  *  SECOND IS RECIEVED.
FA46                       675  * INPUT: X->STRING
FA46                       676  * OUTPUT: STRING SENT TO OUTPUT CONSOLE DRIVER
FA46                       677  *         C=0 NO CTL-X, C=1 CTL-X RECIEVED
FA46                       678  *************************************************
FA46  8D A4                679  ZPDTLP  BSR     SEND            SEND CHARACTER TO DRIVER
FA48  A6 80                680  ZPDTA1  LDA     ,X+             LOAD NEXT CHARACTER
FA4A  81 04                681          CMPA    #EOT            ? EOT
FA4C  26 F8                682          BNE     ZPDTLP          LOOP IF NOT
FA4E                       683  * FALL INTO PAUSE CHECK FUNCTION
FA4E                       684  
FA4E                       685  ********************************************
FA4E                       686  *            [SWI FUNCTION 12]
FA4E                       687  *     PAUSE - RETURN TO TASK DISPATCHING AND CHECK
FA4E                       688  *             FOR FREEZE CONDITION OR CTL-X BREAK
FA4E                       689  *  THIS FUNCTION ENTERS THE TASK PAUSE HANDLER SO
FA4E                       690  *  OPTIONALLY OTHER 6809 PROCESSES MAY GAIN CONTROL.
FA4E                       691  *  UPON RETURN, CHECK FOR A 'FREEZE' CONDITION
FA4E                       692  *  WITH A RESULTING WAIT LOOP, OR CONDITION CODE
FA4E                       693  *  RETURN IF A CONTROL-X IS ENTERED FROM THE INPUT
FA4E                       694  *  HANDLER.
FA4E                       695  * OUTPUT: C=1 IF CTL-X HAS ENTERED, C=0 OTHERWISE
FA4E                       696  ******************************************
FA4E  8D 1E                697  ZPAUSE  BSR     XQPAUS          RELEASE CONTROL AT EVERY LINE
FA50  8D 06                698          BSR     CHKABT          CHECK FOR FREEZE OR ABORT
FA52  1F A9                699          TFR     CC,B            PREPARE TO REPLACE CC
FA54  E7 E4                700          STB     ,S              OVERLAY OLD ONE ON STACK

FA56  20 E1                701          BRA     ZOTCH3          RETURN FROM "SWI"
FA58                       702  
FA58                       703  * CHKABT - SCAN FOR INPUT PAUSE/ABORT DURING OUTPUT
FA58                       704  * OUTPUT: C=0 OK, C=1 ABORT (CTL-X ISSUED)
FA58                       705  * VOLATILE: U,X,D
FA58  8D 18                706  CHKABT  BSR     XQCIDT          ATTEMPT INPUT
FA5A  24 05                707          BCC     CHKRTN          BRANCH NO TO RETURN
FA5C  81 18                708          CMPA    #CAN            ? CTL-X FOR ABORT
FA5E  26 02                709          BNE     CHKWT           BRANCH NO TO PAUSE
FA60  53                   710  CHKSEC  COMB    SET             CARRY
FA61  39                   711  CHKRTN  RTS                     RETURN TO CALLER WITH CC SET
FA62  8D 0A                712  CHKWT   BSR     XQPAUS          PAUSE FOR A MOMENT
FA64  8D 0C                713          BSR     XQCIDT          ? KEY FOR START
FA66  24 FA                714          BCC     CHKWT           LOOP UNTIL RECIEVED
FA68  81 18                715          CMPA    #CAN            ? ABORT SIGNALED FROM WAIT
FA6A  27 F4                716          BEQ     CHKSEC          BRANCH YES
FA6C  4F                   717          CLRA                    SET C=0 FOR NO ABORT
FA6D  39                   718          RTS                     AND RETURN
FA6E                       719  
FA6E                       720  * SAVE MEMORY WITH JUMPS
FA6E  6E 9D E5 78          721  XQPAUS  JMP     [VECTAB+.PAUSE,PCR]     TO PAUSE ROUTINE
FA72  AD 9D E5 62          722  XQCIDT  JSR     [VECTAB+.CIDTA,PCR]     TO INPUT ROUTINE
FA76  84 7F                723          ANDA    #$7F            STRIP PARITY
FA78  39                   724          RTS                     RETURN TO CALLER
FA79                       725  
FA79                       726  ********************************************
FA79                       727  *          NMI DEFAULT INTERRUPT HANDLER
FA79                       728  *  THE NMI HANDLER IS USED FOR TRACING INSTRUCTIONS.
FA79                       729  *  TRACE PRINTOUTS OCCUR ONLY AS LONG AS THE STACK
FA79                       730  *  TRACE LEVEL IS NOT BREACHED BY FALLING BELOW IT.
FA79                       731  *  TRACING CONTINUES UNTIL THE COUNT TURNS ZERO OR
FA79                       732  *  A CTL-X IS ENTERED FROM THE INPUT CONSOLE DEVICE.
FA79                       733  *********************************************
FA79                       734  
FA79  4F 50 2D 04          735  MSHOWP  FCB     'O','P','-',EOT OPCODE PREP
FA7D                       736  
FA7D  8D 42                737  NMIR    BSR     LDDP            LOAD PAGE AND VERIFY STACK
FA7F  0D 8F                738          TST     MISFLG          ? THRU A BREAKPOINT
FA81  26 34                739          BNE     NMICON          BRANCH IF SO TO CONTINUE
FA83  0D 90                740          TST     SWICNT          ? INHIBIT "SWI" DURING TRACE
FA85  2B 29                741          BMI     NMITRC          BRANCH YES
FA87  30 6C                742          LEAX    12,S            OBTAIN USERS STACK POINTER
FA89  9C F8                743          CMPX    SLEVEL          ? TO TRACE HERE
FA8B  25 23                744          BLO     NMITRC          BRANCH IF TOO LOW TO DISPLAY
FA8D  30 8C E9             745          LEAX    MSHOWP,PCR      LOAD OP PREP
FA90  3F                   746          SWI                     SEND TO CONSOLE
FA91  02                   747          FCB     PDATA1          FUNCTION
FA92  09 8E                748          ROL     DELIM           SAVE CARRY BIT
FA94  30 8D E5 01          749          LEAX    LASTOP,PCR      POINT TO LAST OP
FA98  3F                   750          SWI                     SEND OUT AS HEX
FA99  05                   751          FCB     OUT4HS          FUNCTION
FA9A  8D 17                752          BSR     REGPRS          FOLLOW MEMORY WITH REGISTERS
FA9C  25 37                753          BCS     ZBKCMD          BRANCH IF "CANCEL"
FA9E  06 8E                754          ROR     DELIM           RESTORE CARRY BIT
FAA0  25 33                755          BCS     ZBKCMD          BRANCH IF "CANCEL"
FAA2  9E 91                756          LDX     TRACEC          LOAD TRACE COUNT
FAA4  27 2F                757          BEQ     ZBKCMD          IF ZERO TO COMMAND HANDLER
FAA6  30 1F                758          LEAX    -1,X            MINUS ONE
FAA8  9F 91                759          STX     TRACEC          REFRESH
FAAA  27 29                760          BEQ     ZBKCMD          STOP TRACE WHEN ZERO
FAAC  8D AA                761          BSR     CHKABT          ? ABORT THE TRACE
FAAE  25 25                762          BCS     ZBKCMD          BRANCH YES TO COMMAND HANDLER
FAB0  16 03 F7             763  NMITRC  LBRA    CTRCE3          NO, TRACE ANOTHER INSTRUCTION
FAB3                       764  
FAB3  17 01 B9             765  REGPRS  LBSR    REGPRT          PRINT REGISTERS AS FROM COMMAND
FAB6  39                   766          RTS                     RETURN TO CALLER
FAB7                       767  
FAB7                       768  * JUST EXECUTED THRU A BRKPNT.  NOW CONTINUE NORMALLY

FAB7  0F 8F                769  NMICON  CLR     MISFLG          CLEAR THRU FLAG
FAB9  17 02 EB             770          LBSR    ARMBK2          ARM BREAKPOINTS
FABC  3B                   771  RTI     RTI     AND             CONTINUE USERS PROGRAM
FABD                       772  
FABD                       773  * LDDP - SETUP DIRECT PAGE REGISTER, VERIFY STACK.
FABD                       774  * AN INVALID STACK CAUSES A RETURN TO THE COMMAND
FABD                       775  * HANDLER.
FABD                       776  * INPUT: FULLY STACKED REGISTERS FROM AN INTERRUPT
FABD                       777  * OUTPUT: DPR LOADED TO WORK PAGE
FABD                       778  
FABD  3F 07 20 04          779  ERRMSG  FCB     '?',BELL,$20,EOT ERROR RESPONSE
FAC1                       780  
FAC1  E6 8D E4 D8          781  LDDP    LDB     BASEPG,PCR      LOAD DIRECT PAGE HIGH BYTE
FAC5  1F 9B                782          TFR     B,DP            SETUP DIRECT PAGE REGISTER
FAC7  A1 63                783          CMPA    3,S             ? IS STACK VALID
FAC9  27 25                784          BEQ     RTS             YES, RETURN
FACB  10 DE 97             785          LDS     RSTACK          RESET TO INITIAL STACK POINTER
FACE  30 8C EC             786  ERROR   LEAX    ERRMSG,PCR      LOAD ERROR REPORT
FAD1  3F                   787          SWI                     SEND OUT BEFORE REGISTERS
FAD2  03                   788          FCB     PDATA           ON NEXT LINE
FAD3                       789  * FALL INTO BREAKPOINT HANDLER
FAD3                       790  
FAD3                       791  **********************************************
FAD3                       792  *             [SWI FUNCTION 10]
FAD3                       793  *         BREAKPOINT PROGRAM FUNCTION
FAD3                       794  *  PRINT REGISTERS AND GO TO COMMAND HANLER
FAD3                       795  ***********************************************
FAD3  8D DE                796  ZBKPNT  BSR     REGPRS          PRINT OUT REGISTERS
FAD5  16 FE 21             797  ZBKCMD  LBRA    CMDNEP          NOW ENTER COMMAND HANDLER
FAD8                       798  
FAD8                       799  ********************************************
FAD8                       800  *    IRQ, RESERVED, SWI2 AND SWI3 INTERRUPT HANDLERS
FAD8                       801  *  THE DEFAULT HANDLING IS TO CAUSE A BREAKPOINT.
FAD8                       802  ********************************************
FAD8                       803  SWI2R   EQU     *               SWI2 ENTRY
FAD8                       804  SWI3R   EQU     *               SWI3 ENTRY
FAD8                       805  IRQR    EQU     *               IRQ ENTRY
FAD8  8D E7                806  RSRVDR  BSR     LDDP            SET BASE PAGE, VALIDATE STACK
FADA  20 F7                807          BRA     ZBKPNT          FORCE A BREAKPOINT
FADC                       808  
FADC                       809  ******************************************
FADC                       810  *        FIRQ HANDLER
FADC                       811  *  JUST RETURN FOR THE FIRQ INTERRUPT
FADC                       812  ******************************************
FABC                       813  FIRQR   EQU     RTI             IMMEDIATE RETURN
FADC                       814  
FADC                       815  **************************************************
FADC                       816  *      DEFAULT I/O DRIVERS
FADC                       817  **************************************************
FADC                       818  
FADC                       819  * CIDTA - RETURN CONSOLE INPUT CHARACTER
FADC                       820  * OUTPUT: C=0 IF NO DATA READY, C=1 A=CHARACTER
FADC                       821  * U VOLATILE
FADC  DE F0                822  CIDTA   LDU     VECTAB+.ACIA    LOAD ACIA ADDRESS
FADE  A6 C4                823          LDA     ,U              LOAD STATUS REGISTER
FAE0  44                   824          LSRA                    TEST RECEIVER REGISTER FLAG
FAE1  24 02                825          BCC     CIRTN           RETURN IF NOTHING
FAE3  A6 41                826          LDA     1,U             LOAD DATA BYTE
FAE5  39                   827  CIRTN   RTS                     RETURN TO CALLER
FAE6                       828  
FAE6                       829  * CION - INPUT CONSOLE INITIALIZATION
FAE6                       830  * COON - OUTPUT CONSOLE INITIALIZATION
FAE6                       831  * A,X  VOLATILE
FAE6                       832  CION    EQU     *
FAE6  86 03                833  COON    LDA     #3              RESET ACIA CODE
FAE8  9E F0                834          LDX     VECTAB+.ACIA    LOAD ACIA ADDRESS
FAEA  A7 84                835          STA     ,X              STORE INTO STATUS REGISTER
FAEC  86 51                836          LDA     #$51            SET CONTROL

FAEE  A7 84                837          STA     ,X              REGISTER UP
FAF0  39                   838  RTS     RTS                     RETURN TO CALLER
FAF1                       839  
FAF1                       840  * THE FOLLOWING HAVE NO DUTIES TO PERFORM
FAF0                       841  CIOFF   EQU     RTS             CONSOLE INPUT OFF
FAF0                       842  COOFF   EQU     RTS             CONSOLE OUTPUT OFF
FAF1                       843  
FAF1                       844  * CODTA - OUTPUT CHARACTER TO CONSOLE DEVICE
FAF1                       845  * INPUT: A=CHARACTER TO SEND
FAF1                       846  * OUTPUT: CHAR SENT TO TERMINAL WITH PROPER PADDING
FAF1                       847  * ALL REGISTERS TRANSPARENT
FAF1                       848  
FAF1  34 47                849  CODTA   PSHS    U,D,CC          SAVE REGISTERS,WORK BYTE
FAF3  DE F0                850          LDU     VECTAB+.ACIA    ADDRESS ACIA
FAF5  8D 1B                851          BSR     CODTAO          CALL OUTPUT CHAR SUBROTINE
FAF7  81 10                852          CMPA    #DLE            ? DATA LINE ESCAPE
FAF9  27 12                853          BEQ     CODTRT          YES, RETURN
FAFB  D6 F2                854          LDB     VECTAB+.PAD     DEFAULT TO CHAR PAD COUNT
FAFD  81 0D                855          CMPA    #CR             ? CR
FAFF  26 02                856          BNE     CODTPD          BRANCH NO
FB01  D6 F3                857          LDB     VECTAB+.PAD+1   LOAD NEW LINE PAD COUNT
FB03  4F                   858  CODTPD  CLRA                    CREATE NULL
FB04  E7 E4                859          STB     ,S              SAVE COUNT
FB06  8C                   860          FCB     SKIP2           ENTER LOOP
FB07  8D 09                861  CODTLP  BSR     CODTAO          SEND NULL
FB09  6A E4                862          DEC     ,S              ? FINISHED
FB0B  2A FA                863          BPL     CODTLP          NO, CONTINUE WITH MORE
FB0D  35 C7                864  CODTRT  PULS    PC,U,D,CC       RESTORE REGISTERS AND RETURN
FB0F                       865  
FB0F  17 FF 5C             866  CODTAD  LBSR    XQPAUS          TEMPORARY GIVE UP CONTROL
FB12  E6 C4                867  CODTAO  LDB     ,U              LOAD ACIA CONTROL REGISTER
FB14  C5 02                868          BITB    #$02            ? TX REGISTER CLEAR >LSAB FIXME
FB16  26 F7                869          BNE     CODTAD          RELEASE CONTROL IF NOT
FB18  A7 41                870          STA     1,U             STORE INTO DATA REGISTER
FB1A  39                   871          RTS                     RETURN TO CALLER
FB1B                       872  *E
FB1B                       873  
FB1B                       874  * BSON - TURN ON READ/VERIFY/PUNCH MECHANISM
FB1B                       875  * A IS VOLATILE
FB1B                       876  
FB1B  86 11                877  BSON    LDA     #$11            SET READ CODE
FB1D  6D 66                878          TST     6,S             ? READ OR VERIFY
FB1F  26 01                879          BNE     BSON2           BRANCH YES
FB21  4C                   880          INCA                    SET TO WRITE
FB22  3F                   881  BSON2   SWI                     PERFORM OUTPUT
FB23  01                   882          FCB     OUTCH           FUNCTION
FB24  0C 8F                883          INC     MISFLG          SET LOAD IN PROGRESS FLAG
FB26  39                   884          RTS                     RETURN TO CALLER
FB27                       885  
FB27                       886  * BSOFF - TURN OFF READ/VERIFY/PUNCH MECHANISM
FB27                       887  * A,X VOLATILE
FB27  86 14                888  BSOFF   LDA     #$14            TO DC4 - STOP
FB29  3F                   889          SWI                     SEND OUT
FB2A  01                   890          FCB     OUTCH           FUNCTION
FB2B  4A                   891          DECA                    CHANGE TO DC3 (X-OFF)
FB2C  3F                   892          SWI                     SEND OUT
FB2D  01                   893          FCB     OUTCH           FUNCTION
FB2E  0A 8F                894          DEC     MISFLG          CLEAR LOAD IN PROGRESS FLAG
FB30  8E 61 A8             895          LDX     #25000          DELAY 1 SECOND (2MHZ CLOCK)
FB33  30 1F                896  BSOFLP  LEAX    -1,X            COUNT DOWN
FB35  26 FC                897          BNE     BSOFLP          LOOP TILL DONE
FB37  39                   898          RTS                     RETURN TO CALLER
FB38                       899  
FB38                       900  * BSDTA - READ/VERIFY/PUNCH HANDLER
FB38                       901  * INPUT: S+6=CODE BYTE, VERIFY(-1),PUNCH(0),LOAD(1)
FB38                       902  *        S+4=START ADDRESS
FB38                       903  *        S+2=STOP ADDRESS
FB38                       904  *        S+0=RETURN ADDRESS



DUNFIELD 6809 ASSEMBLER: ASSIST09                                     PAGE: 15

FB38                       905  * OUTPUT: Z=1 NORMAL COMPLETION, Z=0 INVALID LOAD/VER
FB38                       906  * REGISTERS ARE VOLATILE
FB38                       907  
FB38  EE 62                908  BSDTA   LDU     2,S             U=TO ADDRESS OR OFFSET
FB3A  6D 66                909          TST     6,S             ? PUNCH
FB3C  27 54                910          BEQ     BSDPUN          BRANCH YES
FB3E                       911  * DURING READ/VERIFY: S+2=MSB ADDRESS SAVE BYTE
FB3E                       912  *                     S+1=BYTE COUNTER
FB3E                       913  *                     S+0=CHECKSUM
FB3E                       914  *                     U HOLDS OFFSET
FB3E  32 7D                915          LEAS    -3,S            ROOM FOR WORK/COUNTER/CHECKSUM
FB40  3F                   916  BSDLD1  SWI                     GET NEXT CHARACTER
FB41  00                   917          FCB     INCHNP          FUNCTION
FB42  81 53                918  BSDLD2  CMPA    #'S'            ? START OF S1/S9
FB44  26 FA                919          BNE     BSDLD1          BRANCH NOT
FB46  3F                   920          SWI                     GET NEXT CHARACTER
FB47  00                   921          FCB     INCHNP          FUNCTION
FB48  81 39                922          CMPA    #'9'            ? HAVE S9
FB4A  27 22                923          BEQ     BSDSRT          YES, RETURN GOOD CODE
FB4C  81 31                924          CMPA    #'1'            ? HAVE NEW RECORD
FB4E  26 F2                925          BNE     BSDLD2          BRANCH IF NOT
FB50  6F E4                926          CLR     ,S              CLEAR CHECKSUM
FB52  8D 21                927          BSR     BYTE            OBTAIN BYTE COUNT
FB54  E7 61                928          STB     1,S             SAVE FOR DECREMENT
FB56                       929  * READ ADDRESS
FB56  8D 1D                930          BSR     BYTE            OBTAIN HIGH VALUE
FB58  E7 62                931          STB     2,S             SAVE IT
FB5A  8D 19                932          BSR     BYTE            OBTAIN LOW VALUE
FB5C  A6 62                933          LDA     2,S             MAKE D=VALUE
FB5E  31 CB                934          LEAY    D,U             Y=ADDRESS+OFFSET
FB60                       935  * STORE TEXT
FB60  8D 13                936  BSDNXT  BSR     BYTE            NEXT BYTE
FB62  27 0C                937          BEQ     BSDEOL          BRANCH IF CHECKSUM
FB64  6D 69                938          TST     9,S             ? VERIFY ONLY
FB66  2B 02                939          BMI     BSDCMP          YES, ONLY COMPARE
FB68  E7 A4                940          STB     ,Y              STORE INTO MEMORY
FB6A  E1 A0                941  BSDCMP  CMPB    ,Y+             ? VALID RAM
FB6C  27 F2                942          BEQ     BSDNXT          YES, CONTINUE READING
FB6E  35 92                943  BSDSRT  PULS    PC,X,A          RETURN WITH Z SET PROPER
FB70                       944  
FB70  4C                   945  BSDEOL  INCA                    ? VALID CHECKSUM
FB71  27 CD                946          BEQ     BSDLD1          BRANCH YES
FB73  20 F9                947          BRA     BSDSRT          RETURN Z=0 INVALID
FB75                       948  
FB75                       949  * BYTE BUILDS 8 BIT VALUE FROM TWO HEX DIGITS IN
FB75  8D 12                950  BYTE    BSR     BYTHEX          OBTAIN FIRST HEX
FB77  C6 10                951          LDB     #16             PREPARE SHIFT
FB79  3D                   952          MUL                     OVER TO A
FB7A  8D 0D                953          BSR     BYTHEX          OBTAIN SECOND HEX
FB7C  34 04                954          PSHS    B               SAVE HIGH HEX
FB7E  AB E0                955          ADDA    ,S+             COMBINE BOTH SIDES
FB80  1F 89                956          TFR     A,B             SEND BACK IN B
FB82  AB 62                957          ADDA    2,S             COMPUTE NEW CHECKSUM
FB84  A7 62                958          STA     2,S             STORE BACK
FB86  6A 63                959          DEC     3,S             DECREMENT BYTE COUNT
FB88  39                   960  BYTRTS  RTS                     RETURN TO CALLER
FB89                       961  
FB89  3F                   962  BYTHEX  SWI                     GET NEXT HEX
FB8A  00                   963          FCB     INCHNP          CHARACTER
FB8B  17 01 D4             964          LBSR    CNVHEX          CONVERT TO HEX
FB8E  27 F8                965          BEQ     BYTRTS          RETURN IF VALID HEX
FB90  35 F2                966          PULS    PC,U,Y,X,A      RETURN TO CALLER WITH Z=0
FB92                       967  
FB92                       968  * PUNCH STACK USE: S+8=TO ADDRESS
FB92                       969  *                  S+6=RETURN ADDRESS
FB92                       970  *                  S+4=SAVED PADDING VALUES
FB92                       971  *                  S+2 FROM ADDRESS
FB92                       972  *                  S+1=FRAME COUNT/CHECKSUM

FB92                       973  *                  S+0=BYTE COUNT
FB92  DE F2                974  BSDPUN  LDU     VECTAB+.PAD     LOAD PADDING VALUES
FB94  AE 64                975          LDX     4,S             X=FROM ADDRESS
FB96  34 56                976          PSHS    U,X,D           CREATE STACK WORK AREA
FB98  CC 00 18             977          LDD     #24             SET A=0, B=24
FB9B  D7 F2                978          STB     VECTAB+.PAD     SETUP 24 CHARACTER PADS
FB9D  3F                   979          SWI                     SEND NULLS OUT
FB9E  01                   980          FCB     OUTCH           FUNCTION
FB9F  C6 04                981          LDB     #4              SETUP NEW LINE PAD TO 4
FBA1  DD F2                982          STD     VECTAB+.PAD     SETUP PUNCH PADDING
FBA3                       983  * CALCULATE SIZE
FBA3  EC 68                984  BSPGO   LDD     8,S             LOAD TO
FBA5  A3 62                985          SUBD    2,S             MINUS FROM=LENGTH
FBA7  10 83 00 18          986          CMPD    #24             ? MORE THAN 23
FBAB  25 02                987          BLO     BSPOK           NO, OK
FBAD  C6 17                988          LDB     #23             FORCE TO 23 MAX
FBAF  5C                   989  BSPOK   INCB                    PREPARE COUNTER
FBB0  E7 E4                990          STB     ,S              STORE BYTE COUNT
FBB2  CB 03                991          ADDB    #3              ADJUST TO FRAME COUNT
FBB4  E7 61                992          STB     1,S             SAVE
FBB6                       993  *PUNCH CR,LF,NULS,S,1
FBB6  30 8C 33             994          LEAX    <BSPSTR,PCR     LOAD START RECORD HEADER
FBB9  3F                   995          SWI                     SEND OUT
FBBA  03                   996          FCB     PDATA           FUNCTION
FBBB                       997  * SEND FRAME COUNT
FBBB  5F                   998          CLRB                    INITIALIZE CHECKSUM
FBBC  30 61                999          LEAX    1,S             POINT TO FRAME COUNT AND ADDR
FBBE  8D 27               1000          BSR     BSPUN2          SEND FRAME COUNT
FBC0                      1001  *DATA ADDRESS
FBC0  8D 25               1002          BSR     BSPUN2          SEND ADDRESS HI
FBC2  8D 23               1003          BSR     BSPUN2          SEND ADDRESS LOW
FBC4                      1004  *PUNCH DATA
FBC4  AE 62               1005          LDX     2,S             LOAD START DATA ADDRESS
FBC6  8D 1F               1006  BSPMRE  BSR     BSPUN2          SEND OUT NEXT BYTE
FBC8  6A E4               1007          DEC     ,S              ? FINAL BYTE
FBCA  26 FA               1008          BNE     BSPMRE          LOOP IF NOT DONE
FBCC  AF 62               1009          STX     2,S             UPDATE FROM ADDRESS VALUE
FBCE                      1010  *PUNCH CHECKSUM
FBCE  53                  1011          COMB    COMPLEMENT
FBCF  E7 61               1012          STB     1,S             STORE FOR SENDOUT
FBD1  30 61               1013          LEAX    1,S             POINT TO IT
FBD3  8D 14               1014          BSR     BSPUNC          SEND OUT AS HEX
FBD5  AE 68               1015          LDX     8,S             LOAD TOP ADDRESS
FBD7  AC 62               1016          CMPX    2,S             ? DONE
FBD9  24 C8               1017          BHS     BSPGO           BRANCH NOT
FBDB  30 8C 11            1018          LEAX    <BSPEOF,PCR     PREPARE END OF FILE
FBDE  3F                  1019          SWI                     SEND OUT STRING
FBDF  03                  1020          FCB     PDATA           FUNCTION
FBE0  EC 64               1021          LDD     4,S             RECOVER PAD COUNTS
FBE2  DD F2               1022          STD     VECTAB+.PAD     RESTORE
FBE4  4F                  1023          CLRA                    SET Z=1 FOR OK RETURN
FBE5  35 D6               1024          PULS    PC,U,X,D        RETURN WITH OK CODE
FBE7                      1025  
FBE7  EB 84               1026  BSPUN2  ADDB    ,X              ADD TO CHECKSUM
FBE9  16 FD ED            1027  BSPUNC  LBRA    ZOUT2H          SEND OUT AS HEX AND RETURN
FBEC                      1028  
FBEC  53 31 04            1029  BSPSTR  FCB     'S','1',EOT     CR,LF,NULLS,S,1
FBEF  53 39 30 33 30 30 + 1030  BSPEOF  FCC     /S9030000FC/    EOF STRING
FBF9  0D 0A 04            1031          FCB     CR,LF,EOT
FBFC                      1032  
FBFC                      1033  * HSDTA - HIGH SPEED PRINT MEMORY
FBFC                      1034  * INPUT: S+4=START ADDRESS
FBFC                      1035  *        S+2=STOP ADDRESS
FBFC                      1036  *        S+0=RETURN ADDRESS
FBFC                      1037  * X,D VOLATILE
FBFC                      1038  
FBFC                      1039  *  SEND TITLE
FBFC  3F                  1040  HSDTA   SWI                     SEND NEW LINE

FBFD  06                  1041          FCB     PCRLF           FUNCTION
FBFE  C6 06               1042          LDB     #6              PREPARE 6 SPACES
FC00  3F                  1043  HSBLNK  SWI                     SEND BLANK
FC01  07                  1044          FCB     SPACE           FUNCTION
FC02  5A                  1045          DECB                    COUNT DOWN
FC03  26 FB               1046          BNE     HSBLNK          LOOP IF MORE
FC05  5F                  1047          CLRB                    SETUP BYTE COUNT
FC06  1F 98               1048  HSHTTL  TFR     B,A             PREPARE FOR CONVERT
FC08  17 FD DB            1049          LBSR    ZOUTHX          CONVERT TO A HEX DIGIT
FC0B  3F                  1050          SWI                     SEND BLANK
FC0C  07                  1051          FCB     SPACE           FUNCTION
FC0D  3F                  1052          SWI                     SEND ANOTHER
FC0E  07                  1053          FCB     SPACE           BLANK
FC0F  5C                  1054          INCB                    UP ANOTHER
FC10  C1 10               1055          CMPB    #$10            ? PAST 'F'
FC12  25 F2               1056          BLO     HSHTTL          LOOP UNTIL SO
FC14  3F                  1057  HSHLNE  SWI                     TO NEXT LINE
FC15  06                  1058          FCB     PCRLF           FUNCTION
FC16  25 2F               1059          BCS     HSDRTN          RETURN IF USER ENTERED CTL-X
FC18  30 64               1060          LEAX    4,S             POINT AT ADDRESS TO CONVERT
FC1A  3F                  1061          SWI                     PRINT OUT ADDRESS
FC1B  05                  1062          FCB     OUT4HS          FUNCTION
FC1C  AE 64               1063          LDX     4,S             LOAD ADDRESS PROPER
FC1E  C6 10               1064          LDB     #16             NEXT SIXTEEN
FC20  3F                  1065  HSHNXT  SWI                     CONVERT BYTE TO HEX AND SEND
FC21  04                  1066          FCB     OUT2HS          FUNCTION
FC22  5A                  1067          DECB                    COUNT DOWN
FC23  26 FB               1068          BNE     HSHNXT          LOOP IF NOT SIXTEENTH
FC25  3F                  1069          SWI                     SEND BLANK
FC26  07                  1070          FCB     SPACE           FUNCTION
FC27  AE 64               1071          LDX     4,S             RELOAD FROM ADDRESS
FC29  C6 10               1072          LDB     #16             COUNT
FC2B  A6 80               1073  HSHCHR  LDA     ,X+             NEXT BYTE
FC2D  2B 04               1074          BMI     HSHDOT          TOO LARGE, TO A DOT
FC2F  81 20               1075          CMPA    #' '            ? LOWER THAN A BLANK
FC31  24 02               1076          BHS     HSHCOK          NO, BRANCH OK
FC33  86 2E               1077  HSHDOT  LDA     #'.'            CONVERT INVALID TO A BLANK
FC35  3F                  1078  HSHCOK  SWI                     SEND CHARACTER
FC36  01                  1079          FCB     OUTCH           FUNCTION
FC37  5A                  1080          DECB                    ? DONE
FC38  26 F1               1081          BNE     HSHCHR          BRANCH NO
FC3A  AC 62               1082          CMPX    2,S             ? PAST LAST ADDRESS
FC3C  24 09               1083          BHS     HSDRTN          QUIT IF SO
FC3E  AF 64               1084          STX     4,S             UPDATE FROM ADDRESS
FC40  A6 65               1085          LDA     5,S             LOAD LOW BYTE ADDRESS
FC42  48                  1086          ASLA    ?               TO SECTION BOUNDRY
FC43  26 CF               1087          BNE     HSHLNE          BRANCH IF NOT
FC45  20 B5               1088          BRA     HSDTA           BRANCH IF SO
FC47  3F                  1089  HSDRTN  SWI                     SEND NEW LINE
FC48  06                  1090          FCB     PCRLF           FUNCTION
FC49  39                  1091          RTS                     RETURN TO CALLER
FC4A                      1092  *F
FC4A                      1093  
FC4A                      1094  ***********************************************
FC4A                      1095  *     A S S I S T 0 9    C O M M A N D S
FC4A                      1096  ***********************************************
FC4A                      1097  
FC4A                      1098  *************REGISTERS - DISPLAY AND CHANGE REGISTERS
FC4A  8D 23               1099  CREG    BSR     REGPRT          PRINT REGISTERS
FC4C  4C                  1100          INCA                    SET FOR CHANGE FUNCTION
FC4D  8D 21               1101          BSR     REGCHG          GO CHANGE, DISPLAY REGISTERS
FC4F  39                  1102          RTS                     RETURN TO COMMAND PROCESSOR
FC50                      1103  
FC50                      1104  ********************************************
FC50                      1105  *      REGPRT - PRINT/CHANGE REGISTERS SUBROUTINE
FC50                      1106  *  WILL ABORT TO 'CMDBAD' IF OVERFLOW DETECTED DURING
FC50                      1107  *  A CHANGE OPERATION.  CHANGE DISPLAYS REGISTERS WHEN
FC50                      1108  *  DONE.

FC50                      1109  * REGISTER MASK LIST CONSISTS OF:
FC50                      1110  *  A) CHARACTERS DENOTING REGISTER
FC50                      1111  *  B) ZERO FOR ONE BYTE, -1 FOR TWO
FC50                      1112  *  C) OFFSET ON STACK TO REGISTER POSITION
FC50                      1113  * INPUT: SP+4=STACKED REGISTERS
FC50                      1114  *        A=0 PRINT, A#0 PRINT AND CHANGE
FC50                      1115  * OUTPUT: (ONLY FOR REGISTER DISPLAY)
FC50                      1116  *         C=1 CONTROL-X ENTERED, C=0 OTHERWISE
FC50                      1117  * VOLATILE: D,X (CHANGE)
FC50                      1118  *           B,X (DISPLAY)
FC50                      1119  *******************************************
FC50  50 43 FF 13         1120  REGMSK  FCB     'P','C',-1,19   PC REG
FC54  41 00 0A            1121          FCB     'A',0,10 A REG
FC57  42 00 0B            1122          FCB     'B',0,11 B REG
FC5A  58 FF 0D            1123          FCB     'X',-1,13 X REG
FC5D  59 FF 0F            1124          FCB     'Y',-1,15 Y REG
FC60  55 FF 11            1125          FCB     'U',-1,17 U REG
FC63  53 FF 01            1126          FCB     'S',-1,1 S REG
FC66  43 43 00 09         1127          FCB     'C','C',0,9     CC REG
FC6A  44 50 00 0C         1128          FCB     'D','P',0,12    DP REG
FC6E  00                  1129          FCB     0               END OF LIST
FC6F                      1130  
FC6F  4F                  1131  REGPRT  CLRA                    SETUP PRINT ONLY FLAG
FC70  30 E8 10            1132  REGCHG  LEAX    4+12,S          READY STACK VALUE
FC73  34 32               1133          PSHS    Y,X,A           SAVE ON STACK WITH OPTION
FC75  31 8C D8            1134          LEAY    REGMSK,PCR      LOAD REGISTER MASK
FC78  EC A0               1135  REGP1   LDD     ,Y+             LOAD NEXT CHAR OR <=0
FC7A  4D                  1136          TSTA    ?               END OF CHARACTERS
FC7B  2F 04               1137          BLE     REGP2           BRANCH NOT CHARACTER
FC7D  3F                  1138          SWI                     SEND TO CONSOLE
FC7E  01                  1139          FCB     OUTCH           FUNCTION BYTE
FC7F  20 F7               1140          BRA     REGP1           CHECK NEXT
FC81  86 2D               1141  REGP2   LDA     #'-'            READY '-'
FC83  3F                  1142          SWI                     SEND OUT
FC84  01                  1143          FCB     OUTCH           WITH OUTCH
FC85  30 E5               1144          LEAX    B,S             X->REGISTER TO PRINT
FC87  6D E4               1145          TST     ,S              ? CHANGE OPTION
FC89  26 12               1146          BNE     REGCNG          BRANCH YES
FC8B  6D 3F               1147          TST     -1,Y            ? ONE OR TWO BYTES
FC8D  27 03               1148          BEQ     REGP3           BRANCH ZERO MEANS ONE
FC8F  3F                  1149          SWI                     PERFORM WORD HEX
FC90  05                  1150          FCB     OUT4HS          FUNCTION
FC91  8C                  1151          FCB     SKIP2           SKIP BYTE PRINT
FC92  3F                  1152  REGP3   SWI                     PERFORM BYTE HEX
FC93  04                  1153          FCB     OUT2HS          FUNCTION
FC94  EC A0               1154  REG4    LDD     ,Y+             TO FRONT OF NEXT ENTRY
FC96  5D                  1155          TSTB    ?               END OF ENTRIES
FC97  26 DF               1156          BNE     REGP1           LOOP IF MORE
FC99  3F                  1157          SWI                     FORCE NEW LINE
FC9A  06                  1158          FCB     PCRLF           FUNCTION
FC9B  35 B2               1159  REGRTN  PULS    PC,Y,X,A        RESTORE STACK AND RETURN
FC9D                      1160  
FC9D  8D 40               1161  REGCNG  BSR     BLDNNB          INPUT BINARY NUMBER
FC9F  27 10               1162          BEQ     REGNXC          IF CHANGE THEN JUMP
FCA1  81 0D               1163          CMPA    #CR             ? NO MORE DESIRED
FCA3  27 1E               1164          BEQ     REGAGN          BRANCH NOPE
FCA5  E6 3F               1165          LDB     -1,Y            LOAD SIZE FLAG
FCA7  5A                  1166          DECB                    MINUS ONE
FCA8  50                  1167          NEGB                    MAKE POSITIVE
FCA9  58                  1168          ASLB    TIMES           TWO (=2 OR =4)
FCAA  3F                  1169  REGSKP  SWI                     PERFORM SPACES
FCAB  07                  1170          FCB     SPACE           FUNCTION
FCAC  5A                  1171          DECB
FCAD  26 FB               1172          BNE     REGSKP          LOOP IF MORE
FCAF  20 E3               1173          BRA     REG4            CONTINUE WITH NEXT REGISTER
FCB1  A7 E4               1174  REGNXC  STA     ,S              SAVE DELIMITER IN OPTION
FCB3                      1175  *                     (ALWAYS > 0)
FCB3  DC 9B               1176          LDD     NUMBER          OBTAIN BINARY RESULT

FCB5  6D 3F               1177          TST     -1,Y            ? TWO BYTES WORTH
FCB7  26 02               1178          BNE     REGTWO          BRANCH YES
FCB9  A6 82               1179          LDA     ,-X             SETUP FOR TWO
FCBB  ED 84               1180  REGTWO  STD     ,X              STORE IN NEW VALUE
FCBD  A6 E4               1181          LDA     ,S              RECOVER DELIMITER
FCBF  81 0D               1182          CMPA    #CR             ? END OF CHANGES
FCC1  26 D1               1183          BNE     REG4            NO, KEEP ON TRUCK'N
FCC3                      1184  * MOVE STACKED DATA TO NEW STACK IN CASE STACK
FCC3                      1185  * POINTER HAS CHANGED
FCC3  30 8D E2 8A         1186  REGAGN  LEAX    TSTACK,PCR      LOAD TEMP AREA
FCC7  C6 15               1187          LDB     #21             LOAD COUNT
FCC9  35 02               1188  REGTF1  PULS    A               NEXT BYTE
FCCB  A7 80               1189          STA     ,X+             STORE INTO TEMP
FCCD  5A                  1190          DECB                    COUNT DOWN
FCCE  26 F9               1191          BNE     REGTF1          LOOP IF MORE
FCD0  10 EE 88 EC         1192          LDS     -20,X           LOAD NEW STACK POINTER
FCD4  C6 15               1193          LDB     #21             LOAD COUNT AGAIN
FCD6  A6 82               1194  REGTF2  LDA     ,-X             NEXT TO STORE
FCD8  34 02               1195          PSHS    A               BACK ONTO NEW STACK
FCDA  5A                  1196          DECB                    COUNT DOWN
FCDB  26 F9               1197          BNE     REGTF2          LOOP IF MORE
FCDD  20 BC               1198          BRA     REGRTN          GO RESTART COMMAND
FCDF                      1199  
FCDF                      1200  *********************************************
FCDF                      1201  *  BLDNUM - BUILDS BINARY VALUE FROM INPUT HEX
FCDF                      1202  *  THE ACTIVE EXPRESSION HANDLER IS USED.
FCDF                      1203  * INPUT: S=RETURN ADDRESS
FCDF                      1204  * OUTPUT: A=DELIMITER WHICH TERMINATED VALUE
FCDF                      1205  *                            (IF DELM NOT ZERO)
FCDF                      1206  *         "NUMBER"=WORD BINARY RESULT
FCDF                      1207  *         Z=1 IF INPUT RECIEVED, Z=0 IF NO HEX RECIEVED
FCDF                      1208  *  REGISTERS ARE TRANSPARENT
FCDF                      1209  **********************************************
FCDF                      1210  
FCDF                      1211  * EXECUTE SINGLE OR EXTENDED ROM EXPRESSION HANDLER
FCDF                      1212  *
FCDF                      1213  * THE FLAG "DELIM" IS USED AS FOLLOWS:
FCDF                      1214  *   DELIM=0  NO LEADING BLANKS, NO FORCED TERMINATOR
FCDF                      1215  *   DELIM=CHR  ACCEPT LEADING 'CHR'S, FORCED TERMINATOR
FCDF  4F                  1216  BLDNNB  CLRA                    NO DYNAMIC DELIMITER
FCE0  8C                  1217          FCB     SKIP2           SKIP NEXT INSTRUCTION
FCE1                      1218  * BUILD WITH LEADING BLANKS
FCE1  86 20               1219  BLDNUM  LDA     #' '            ALLOW LEADING BLANKS
FCE3  97 8E               1220          STA     DELIM           STORE AS DELIMITER
FCE5  6E 9D E3 03         1221          JMP     [VECTAB+.EXPAN,PCR]     TO EXP ANALYZER
FCE9                      1222  
FCE9                      1223  * THIS IS THE DEFAULT SINGLE ROM ANALYZER. WE ACCEPT:
FCE9                      1224  *    1) HEX INPUT
FCE9                      1225  *    2) 'M' FOR LAST MEMORY EXAMINE ADDRESS
FCE9                      1226  *    3) 'P' FOR PROGRAM COUNTER ADDRESS
FCE9                      1227  *    4) 'W' FOR WINDOW VALUE
FCE9                      1228  *    5) '@' FOR INDIRECT VALUE
FCE9  34 14               1229  EXP1    PSHS    X,B             SAVE REGISTERS
FCEB  8D 5C               1230  EXPDLM  BSR     BLDHXI          CLEAR NUMBER, CHECK FIRST CHAR
FCED  27 18               1231          BEQ     EXP2            IF HEX DIGIT CONTINUE BUILDING
FCEF                      1232  * SKIP BLANKS IF DESIRED
FCEF  91 8E               1233          CMPA    DELIM           ? CORRECT DELIMITER
FCF1  27 F8               1234          BEQ     EXPDLM          YES, IGNORE IT
FCF3                      1235  * TEST FOR M OR P
FCF3  9E 9E               1236          LDX     ADDR            DEFAULT FOR 'M'
FCF5  81 4D               1237          CMPA    #'M'            ? MEMORY EXAMINE ADDR WANTED
FCF7  27 16               1238          BEQ     EXPTDL          BRANCH IF SO
FCF9  9E 93               1239          LDX     PCNTER          DEFAULT FOR 'P'
FCFB  81 50               1240          CMPA    #'P'            ? LAST PROGRAM COUNTER WANTED
FCFD  27 10               1241          BEQ     EXPTDL          BRANCH IF SO
FCFF  9E A0               1242          LDX     WINDOW          DEFAULT TO WINDOW
FD01  81 57               1243          CMPA    #'W'            ? WINDOW WANTED
FD03  27 0A               1244          BEQ     EXPTDL

FD05  35 94               1245  EXPRTN  PULS    PC,X,B          RETURN AND RESTORE REGISTERS
FD07                      1246  * GOT HEX, NOW CONTINUE BUILDING
FD07  8D 44               1247  EXP2    BSR     BLDHEX          COMPUTE NEXT DIGIT
FD09  27 FC               1248          BEQ     EXP2            CONTINUE IF MORE
FD0B  20 0A               1249          BRA     EXPCDL          SEARCH FOR +/-
FD0D                      1250  * STORE VALUE AND CHECK IF NEED DELIMITER
FD0D  AE 84               1251  EXPTDI  LDX     ,X              INDIRECTION DESIRED
FD0F  9F 9B               1252  EXPTDL  STX     NUMBER          STORE RESULT
FD11  0D 8E               1253          TST     DELIM           ? TO FORCE A DELIMITER
FD13  27 F0               1254          BEQ     EXPRTN          RETURN IF NOT WITH VALUE
FD15  8D 62               1255          BSR     READ            OBTAIN NEXT CHARACTER
FD17                      1256  * TEST FOR + OR -
FD17  9E 9B               1257  EXPCDL  LDX     NUMBER          LOAD LAST VALUE
FD19  81 2B               1258          CMPA    #'+'            ? ADD OPERATOR
FD1B  26 0E               1259          BNE     EXPCHM          BRANCH NOT
FD1D  8D 23               1260          BSR     EXPTRM          COMPUTE NEXT TERM
FD1F  34 02               1261          PSHS    A               SAVE DELIMITER
FD21  DC 9B               1262          LDD     NUMBER          LOAD NEW TERM
FD23  30 8B               1263  EXPADD  LEAX    D,X             ADD TO X
FD25  9F 9B               1264          STX     NUMBER          STORE AS NEW RESULT
FD27  35 02               1265          PULS    A               RESTORE DELIMITER
FD29  20 EC               1266          BRA     EXPCDL          NOW TEST IT
FD2B  81 2D               1267  EXPCHM  CMPA    #'-'            ? SUBTRACT OPERATOR
FD2D  27 07               1268          BEQ     EXPSUB          BRANCH IF SO
FD2F  81 40               1269          CMPA    #'@'            ? INDIRECTION DESIRED
FD31  27 DA               1270          BEQ     EXPTDI          BRANCH IF SO
FD33  5F                  1271          CLRB                    SET DELIMITER RETURN
FD34  20 CF               1272          BRA     EXPRTN          AND RETURN TO CALLER
FD36  8D 0A               1273  EXPSUB  BSR     EXPTRM          OBTAIN NEXT TERM
FD38  34 02               1274          PSHS    A               SAVE DELIMITER
FD3A  DC 9B               1275          LDD     NUMBER          LOAD UP NEXT TERM
FD3C  40                  1276          NEGA                    NEGATE A
FD3D  50                  1277          NEGB                    NEGATE B
FD3E  82 00               1278          SBCA    #0              CORRECT FOR A
FD40  20 E1               1279          BRA     EXPADD          GO ADD TO EXPRESION
FD42                      1280  * COMPUTE NEXT EXPRESSION TERM
FD42                      1281  * OUTPUT: X=OLD VALUE
FD42                      1282  *         'NUMBER'=NEXT TERM
FD42  8D 9D               1283  EXPTRM  BSR     BLDNUM          OBTAIN NEXT VALUE
FD44  27 32               1284          BEQ     CNVRTS          RETURN IF VALID NUMBER
FD46  16 FC 13            1285  BLDBAD  LBRA    CMDBAD          ABORT COMMAND IF INVALID
FD49                      1286  
FD49                      1287  *********************************************
FD49                      1288  *  BUILD BINARY VALUE USING INPUT CHARACTERS.
FD49                      1289  * INPUT: A=ASCII HEX VALUE OR DELIMITER
FD49                      1290  *        SP+0=RETURN ADDRESS
FD49                      1291  *        SP+2=16 BIT RESULT AREA
FD49                      1292  * OUTPUT: Z=1 A=BINARY VALUE
FD49                      1293  *         Z=0 IF INVALID HEX CHARACTER (A UNCHANGED)
FD49                      1294  * VOLATILE: D
FD49                      1295  ****************************************
FD49  0F 9B               1296  BLDHXI  CLR     NUMBER          CLEAR NUMBER
FD4B  0F 9C               1297          CLR     NUMBER+1        CLEAR NUMBER
FD4D  8D 2A               1298  BLDHEX  BSR     READ            GET INPUT CHARACTER
FD4F  8D 11               1299  BLDHXC  BSR     CNVHEX          CONVERT AND TEST CHARACTER
FD51  26 25               1300          BNE     CNVRTS          RETURN IF NOT A NUMBER
FD53  C6 10               1301          LDB     #16             PREPARE SHIFT
FD55  3D                  1302          MUL                     BY FOUR PLACES
FD56  86 04               1303          LDA     #4              ROTATE BINARY INTO VALUE
FD58  58                  1304  BLDSHF  ASLB    OBTAIN          NEXT BIT
FD59  09 9C               1305          ROL     NUMBER+1        INTO LOW BYTE
FD5B  09 9B               1306          ROL     NUMBER          INTO HI BYTE
FD5D  4A                  1307          DECA                    COUNT DOWN
FD5E  26 F8               1308          BNE     BLDSHF          BRANCH IF MORE TO DO
FD60  20 14               1309          BRA     CNVOK           SET GOOD RETURN CODE
FD62                      1310  
FD62                      1311  ****************************************
FD62                      1312  * CONVERT ASCII CHARACTER TO BINARY BYTE

FD62                      1313  * INPUT: A=ASCII
FD62                      1314  * OUTPUT: Z=1 A=BINARY VALUE
FD62                      1315  *         Z=0 IF INVALID
FD62                      1316  * ALL REGISTERS TRANSPARENT
FD62                      1317  * (A UNALTERED IF INVALID HEX)
FD62                      1318  **************************************
FD62  81 30               1319  CNVHEX  CMPA    #'0'            ? LOWER THAN A ZERO
FD64  25 12               1320          BLO     CNVRTS          BRANCH NOT VALUE
FD66  81 39               1321          CMPA    #'9'            ? POSSIBLE A-F
FD68  2F 0A               1322          BLE     CNVGOT          BRANCH NO TO ACCEPT
FD6A  81 41               1323          CMPA    #'A'            ? LESS THEN TEN
FD6C  25 0A               1324          BLO     CNVRTS          RETURN IF MINUS (INVALID)
FD6E  81 46               1325          CMPA    #'F'            ? NOT TOO LARGE
FD70  22 06               1326          BHI     CNVRTS          NO, RETURN TOO LARGE
FD72  80 07               1327          SUBA    #7              DOWN TO BINARY
FD74  84 0F               1328  CNVGOT  ANDA    #$0F            CLEAR HIGH HEX
FD76  1A 04               1329  CNVOK   ORCC    #4              FORCE ZERO ON FOR VALID HEX
FD78  39                  1330  CNVRTS  RTS                     RETURN TO CALLER
FD79                      1331  
FD79                      1332  * GET INPUT CHAR, ABORT COMMAND IF CONTROL-X (CANCEL)
FD79  3F                  1333  READ    SWI                     GET NEXT CHARACTER
FD7A  00                  1334          FCB     INCHNP          FUNCTION
FD7B  81 18               1335          CMPA    #CAN            ? ABORT COMMAND
FD7D  27 C7               1336          BEQ     BLDBAD          BRANCH TO ABORT IF SO
FD7F  39                  1337          RTS                     RETURN TO CALLER
FD80                      1338  *G
FD80                      1339  
FD80                      1340  ***************GO - START PROGRAM EXECUTION
FD80  8D 01               1341  CGO     BSR     GOADDR          BUILD ADDRESS IF NEEDED
FD82  3B                  1342          RTI     START           EXECUTING
FD83                      1343  
FD83                      1344  * FIND OPTIONAL NEW PROGRAM COUNTER. ALSO ARM THE
FD83                      1345  * BREAKPOINTS.
FD83  35 30               1346  GOADDR  PULS    Y,X             RECOVER RETURN ADDRESS
FD85  34 10               1347          PSHS    X               STORE RETURN BACK
FD87  26 19               1348          BNE     GONDFT          IF NO CARRIAGE RETURN THEN NEW PC
FD89                      1349  * DEFAULT PROGRAM COUNTER, SO FALL THROUGH IF
FD89                      1350  * IMMEDIATE BREAKPOINT.
FD89  17 01 B6            1351          LBSR    CBKLDR          SEARCH BREAKPOINTS
FD8C  AE 6C               1352          LDX     12,S            LOAD PROGRAM COUNTER
FD8E  5A                  1353  ARMBLP  DECB                    COUNT DOWN
FD8F  2B 16               1354          BMI     ARMBK2          DONE, NONE TO SINGLE TRACE
FD91  A6 30               1355          LDA     -NUMBKP*2,Y     PRE-FETCH OPCODE
FD93  AC A1               1356          CMPX    ,Y++            ? IS THIS A BREAKPOINT
FD95  26 F7               1357          BNE     ARMBLP          LOOP IF NOT
FD97  81 3F               1358          CMPA    #$3F            ? SWI BREAKPOINTED
FD99  26 02               1359          BNE     ARMNSW          NO, SKIP SETTING OF PASS FLAG
FD9B  97 FB               1360          STA     SWIBFL          SHOW UPCOMMING SWI NOT BRKPNT
FD9D  0C 8F               1361  ARMNSW  INC     MISFLG          FLAG THRU A BREAKPOINT
FD9F  16 01 06            1362          LBRA    CDOT            DO SINGLE TRACE W/O BREAKPOINTS
FDA2                      1363  * OBTAIN NEW PROGRAM COUNTER
FDA2  17 00 BB            1364  GONDFT  LBSR    CDNUM           OBTAIN NEW PROGRAM COUNTER
FDA5  ED 6C               1365          STD     12,S            STORE INTO STACK
FDA7  17 01 98            1366  ARMBK2  LBSR    CBKLDR          OBTAIN TABLE
FDAA  00 FA               1367          NEG     BKPTCT          COMPLEMENT TO SHOW ARMED
FDAC  5A                  1368  ARMLOP  DECB                    ? DONE
FDAD  2B C9               1369          BMI     CNVRTS          RETURN WHEN DONE
FDAF  A6 B4               1370          LDA     [,Y]            LOAD OPCODE
FDB1  A7 30               1371          STA     -NUMBKP*2,Y     STORE INTO OPCODE TABLE
FDB3  86 3F               1372          LDA     #$3F            READY "SWI" OPCODE
FDB5  A7 B1               1373          STA     [,Y++]          STORE AND MOVE UP TABLE
FDB7  20 F3               1374          BRA     ARMLOP          AND CONTINUE
FDB9                      1375  
FDB9                      1376  *******************CALL - CALL ADDRESS AS SUBROUTINE
FDB9  8D C8               1377  CCALL   BSR     GOADDR          FETCH ADDRESS IF NEEDED
FDBB  35 7F               1378          PULS    U,Y,X,DP,D,CC   RESTORE USERS REGISTERS
FDBD  AD F1               1379          JSR     [,S++]          CALL USER SUBROUTINE
FDBF  3F                  1380  CGOBRK  SWI                     PERFORM BREAKPOINT

FDC0  0A                  1381          FCB     BRKPT           FUNCTION
FDC1  20 FC               1382          BRA     CGOBRK          LOOP UNTIL USER CHANGES PC
FDC3                      1383  
FDC3                      1384  ****************MEMORY - DISPLAY/CHANGE MEMORY
FDC3                      1385  * CMEMN AND CMPADP ARE DIRECT ENTRY POINTS FROM
FDC3                      1386  * THE COMMAND HANDLER FOR QUICK COMMANDS
FDC3  17 00 9A            1387  CMEM    LBSR    CDNUM           OBTAIN ADDRESS
FDC6  DD 9E               1388  CMEMN   STD     ADDR            STORE DEFAULT
FDC8  9E 9E               1389  CMEM2   LDX     ADDR            LOAD POINTER
FDCA  17 FC 0C            1390          LBSR    ZOUT2H          SEND OUT HEX VALUE OF BYTE
FDCD  86 2D               1391          LDA     #'-'            LOAD DELIMITER
FDCF  3F                  1392          SWI                     SEND OUT
FDD0  01                  1393          FCB     OUTCH           FUNCTION
FDD1  17 FF 0B            1394  CMEM4   LBSR    BLDNNB          OBTAIN NEW BYTE VALUE
FDD4  27 0A               1395          BEQ     CMENUM          BRANCH IF NUMBER
FDD6                      1396  * COMA - SKIP BYTE
FDD6  81 2C               1397          CMPA    #','            ? COMMA
FDD8  26 0E               1398          BNE     CMNOTC          BRANCH NOT
FDDA  9F 9E               1399          STX     ADDR            UPDATE POINTER
FDDC  30 01               1400          LEAX    1,X             TO NEXT BYTE
FDDE  20 F1               1401          BRA     CMEM4           AND INPUT IT
FDE0  D6 9C               1402  CMENUM  LDB     NUMBER+1        LOAD LOW BYTE VALUE
FDE2  8D 47               1403          BSR     MUPDAT          GO OVERLAY MEMORY BYTE
FDE4  81 2C               1404          CMPA    #','            ? CONTINUE WITH NO DISPLAY
FDE6  27 E9               1405          BEQ     CMEM4           BRANCH YES
FDE8                      1406  * QUOTED STRING
FDE8  81 27               1407  CMNOTC  CMPA    #$27            ? QUOTED STRING
FDEA  26 0C               1408          BNE     CMNOTQ          BRANCH NO
FDEC  8D 8B               1409  CMESTR  BSR     READ            OBTAIN NEXT CHARACTER
FDEE  81 27               1410          CMPA    #$27            ? END OF QUOTED STRING
FDF0  27 0C               1411          BEQ     CMSPCE          YES, QUIT STRING MODE
FDF2  1F 89               1412          TFR     A,B             TO B FOR SUBROUTINE
FDF4  8D 35               1413          BSR     MUPDAT          GO UPDATE BYTE
FDF6  20 F4               1414          BRA     CMESTR          GET NEXT CHARACTER
FDF8                      1415  * BLANK - NEXT BYTE
FDF8  81 20               1416  CMNOTQ  CMPA    #$20            ? BLANK FOR NEXT BYTE
FDFA  26 06               1417          BNE     CMNOTB          BRANCH NOT
FDFC  9F 9E               1418          STX     ADDR            UPDATE POINTER
FDFE  3F                  1419  CMSPCE  SWI                     GIVE SPACE
FDFF  07                  1420          FCB     SPACE           FUNCTION
FE00  20 C6               1421          BRA     CMEM2           NOW PROMPT FOR NEXT
FE02                      1422  * LINE FEED - NEXT BYTE WITH ADDRESS
FE02  81 0A               1423  CMNOTB  CMPA    #LF             ? LINE FEED FOR NEXT BYTE
FE04  26 08               1424          BNE     CMNOTL          BRANCH NO
FE06  86 0D               1425          LDA     #CR             GIVE CARRIAGE RETURN
FE08  3F                  1426          SWI                     TO CONSOLE
FE09  01                  1427          FCB     OUTCH           HANDLER
FE0A  9F 9E               1428          STX     ADDR            STORE NEXT ADDRESS
FE0C  20 0A               1429          BRA     CMPADP          BRANCH TO SHOW
FE0E                      1430  * UP ARROW - PREVIOUS BYTE AND ADDRESS
FE0E  81 5E               1431  CMNOTL  CMPA    #'^'            ? UP ARROW FOR PREVIOUS BYTE
FE10  26 0A               1432          BNE     CMNOTU          BRANCH NOT
FE12  30 1E               1433          LEAX    -2,X            DOWN TO PREVIOUS BYTE
FE14  9F 9E               1434          STX     ADDR            STORE NEW POINTER
FE16  3F                  1435  CMPADS  SWI                     FORCE NEW LINE
FE17  06                  1436          FCB     PCRLF           FUNCTION
FE18  8D 07               1437  CMPADP  BSR     PRTADR          GO PRINT ITS VALUE
FE1A  20 AC               1438          BRA     CMEM2           THEN PROMPT FOR INPUT
FE1C                      1439  * SLASH - NEXT BYTE WITH ADDRESS
FE1C  81 2F               1440  CMNOTU  CMPA    #'/'            ? SLASH FOR CURRENT DISPLAY
FE1E  27 F6               1441          BEQ     CMPADS          YES, SEND ADDRESS
FE20  39                  1442          RTS                     RETURN FROM COMMAND
FE21                      1443  
FE21                      1444  * PRINT CURRENT ADDRESS
FE21  9E 9E               1445  PRTADR  LDX     ADDR            LOAD POINTER VALUE
FE23  34 10               1446          PSHS    X               SAVE X ON STACK
FE25  30 E4               1447          LEAX    ,S              POINT TO IT FOR DISPLAY
FE27  3F                  1448          SWI                     DISPLAY POINTER IN HEX

FE28  05                  1449          FCB     OUT4HS          FUNCTION
FE29  35 90               1450          PULS    PC,X            RECOVER POINTER AND RETURN
FE2B                      1451  
FE2B                      1452  * UPDATE BYTE
FE2B  9E 9E               1453  MUPDAT  LDX     ADDR            LOAD NEXT BYTE POINTER
FE2D  E7 80               1454          STB     ,X+             STORE AND INCREMENT X
FE2F  E1 1F               1455          CMPB    -1,X            ? SUCCESFULL STORE
FE31  26 03               1456          BNE     MUPBAD          BRANCH FOR '?' IF NOT
FE33  9F 9E               1457          STX     ADDR            STORE NEW POINTER VALUE
FE35  39                  1458          RTS                     BACK TO CALLER
FE36  34 02               1459  MUPBAD  PSHS    A               SAVE A REGISTER
FE38  86 3F               1460          LDA     #'?'            SHOW INVALID
FE3A  3F                  1461          SWI                     SEND OUT
FE3B  01                  1462          FCB     OUTCH           FUNCTION
FE3C  35 82               1463          PULS    PC,A            RETURN TO CALLER
FE3E                      1464  
FE3E                      1465  ********************WINDOW  -  SET WINDOW VALUE
FE3E  8D 20               1466  CWINDO  BSR     CDNUM           OBTAIN WINDOW VALUE
FE40  DD A0               1467          STD     WINDOW          STORE IT IN
FE42  39                  1468          RTS                     END COMMAND
FE43                      1469  
FE43                      1470  ******************DISPLAY - HIGH SPEED DISPLAY MEMORY
FE43  8D 1B               1471  CDISP   BSR     CDNUM           FETCH ADDRESS
FE45  C4 F0               1472          ANDB    #$F0            FORCE TO 16 BOUNDRY
FE47  1F 02               1473          TFR     D,Y             SAVE IN Y
FE49  30 2F               1474          LEAX    15,Y            DEFAULT LENGTH
FE4B  25 04               1475          BCS     CDISPS          BRANCH IF END OF INPUT
FE4D  8D 11               1476          BSR     CDNUM           OBTAIN COUNT
FE4F  30 AB               1477          LEAX    D,Y             ASSUME COUNT, COMPUTE END ADDR
FE51  34 30               1478  CDISPS  PSHS    Y,X             SETUP PARAMETERS FOR HSDATA
FE53  10 A3 62            1479          CMPD    2,S             ? WAS IT COUNT
FE56  23 02               1480          BLS     CDCNT           BRANCH YES
FE58  ED E4               1481          STD     ,S              STORE HIGH ADDRESS
FE5A  AD 9D E1 84         1482  CDCNT   JSR     [VECTAB+.HSDTA,PCR]     CALL PRINT ROUTINE
FE5E  35 E0               1483          PULS    PC,U,Y          CLEAN STACK AND END COMMAND
FE60                      1484  
FE60                      1485  * OBTAIN NUMBER - ABORT IF NONE
FE60                      1486  * ONLY DELIMITERS OF CR, BLANK, OR '/' ARE ACCEPTED
FE60                      1487  * OUTPUT: D=VALUE, C=1 IF CARRIAGE RETURN DELMITER,
FE60                      1488  *                                  ELSE C=0
FE60  17 FE 7E            1489  CDNUM   LBSR    BLDNUM          OBTAIN NUMBER
FE63  26 09               1490          BNE     CDBADN          BRANCH IF INVALID
FE65  81 2F               1491          CMPA    #'/'            ? VALID DELIMITER
FE67  22 05               1492          BHI     CDBADN          BRANCH IF NOT FOR ERROR
FE69  81 0E               1493          CMPA    #CR+1           LEAVE COMPARE FOR CARRIAGE RET
FE6B  DC 9B               1494          LDD     NUMBER          LOAD NUMBER
FE6D  39                  1495          RTS                     RETURN WITH COMPARE
FE6E  16 FA EB            1496  CDBADN  LBRA    CMDBAD          RETURN TO ERROR MECHANISM
FE71                      1497  
FE71                      1498  *****************PUNCH - PUNCH MEMORY IN S1-S9 FORMAT
FE71  8D ED               1499  CPUNCH  BSR     CDNUM           OBTAIN START ADDRESS
FE73  1F 02               1500          TFR     D,Y             SAVE IN Y
FE75  8D E9               1501          BSR     CDNUM           OBTAIN END ADDRESS
FE77  6F E2               1502          CLR     ,-S             SETUP PUNCH FUNCTION CODE
FE79  34 26               1503          PSHS    Y,D             STORE VALUES ON STACK
FE7B  AD 9D E1 65         1504  CCALBS  JSR     [VECTAB+.BSON,PCR]      INITIALIZE HANDLER
FE7F  AD 9D E1 63         1505          JSR     [VECTAB+.BSDTA,PCR]     PERFORM FUNCTION
FE83  34 01               1506          PSHS    CC              SAVE RETURN CODE
FE85  AD 9D E1 5F         1507          JSR     [VECTAB+.BSOFF,PCR]     TURN OFF HANDLER
FE89  35 01               1508          PULS    CC              OBTAIN CONDITION CODE SAVED
FE8B  26 E1               1509          BNE     CDBADN          BRANCH IF ERROR
FE8D  35 B2               1510          PULS    PC,Y,X,A        RETURN FROM COMMAND
FE8F                      1511  
FE8F                      1512  *****************LOAD - LOAD MEMORY FROM S1-S9 FORMAT
FE8F  8D 01               1513  CLOAD   BSR     CLVOFS          CALL SETUP AND PASS CODE
FE91  01                  1514          FCB     1               LOAD FUNCTION CODE FOR PACKET
FE92                      1515  
FE92  33 F1               1516  CLVOFS  LEAU    [,S++]          LOAD CODE IN HIGH BYTE OF U

FE94  33 D4               1517          LEAU    [,U]            NOT CHANGING CC AND RESTORE S
FE96  27 03               1518          BEQ     CLVDFT          BRANCH IF CARRIAGE RETURN NEXT
FE98  8D C6               1519          BSR     CDNUM           OBTAIN OFFSET
FE9A  8C                  1520          FCB     SKIP2           SKIP DEFAULT OFFSET
FE9B  4F                  1521  CLVDFT  CLRA                    CREATE ZERO OFFSET
FE9C  5F                  1522          CLRB                    AS DEFAULT
FE9D  34 4E               1523          PSHS    U,DP,D          SETUP CODE, NULL WORD, OFFSET
FE9F  20 DA               1524          BRA     CCALBS          ENTER CALL TO BS ROUTINES
FEA1                      1525  
FEA1                      1526  ******************VERIFY - COMPARE MEMORY WITH FILES
FEA1  8D EF               1527  CVER    BSR     CLVOFS          COMPUTE OFFSET IF ANY
FEA3  FF                  1528          FCB     -1              VERIFY FNCTN CODE FOR PACKET
FEA4                      1529  
FEA4                      1530  *******************TRACE - TRACE INSTRUCTIONS
FEA4                      1531  ******************* . - SINGLE STEP TRACE
FEA4  8D BA               1532  CTRACE  BSR     CDNUM           OBTAIN TRACE COUNT
FEA6  DD 91               1533          STD     TRACEC          STORE COUNT
FEA8  32 62               1534  CDOT    LEAS    2,S             RID COMMAND RETURN FROM STACK
FEAA  EE F8 0A            1535  CTRCE3  LDU     [10,S]          LOAD OPCODE TO EXECUTE
FEAD  DF 99               1536          STU     LASTOP          STORE FOR TRACE INTERRUPT
FEAF  DE F6               1537          LDU     VECTAB+.PTM     LOAD PTM ADDRESS
FEB1  CC 07 01            1538          LDD     #$0701          7,1 CYCLES DOWN+CYCLES UP
FEB4  ED 42               1539          STD     PTMTM1-PTM,U    START NMI TIMEOUT
FEB6  3B                  1540          RTI     RETURN          FOR ONE INSTRUCTION
FEB7                      1541  
FEB7                      1542  *************NULLS  -  SET NEW LINE AND CHAR PADDING
FEB7  8D A7               1543  CNULLS  BSR     CDNUM           OBTAIN NEW LINE PAD
FEB9  DD F2               1544          STD     VECTAB+.PAD     RESET VALUES
FEBB  39                  1545          RTS                     END COMMAND
FEBC                      1546  
FEBC                      1547  ******************STLEVEL - SET STACK TRACE LEVEL
FEBC  27 05               1548  CSTLEV  BEQ     STLDFT          TAKE DEFAULT
FEBE  8D A0               1549          BSR     CDNUM           OBTAIN NEW STACK LEVEL
FEC0  DD F8               1550          STD     SLEVEL          STORE NEW ENTRY
FEC2  39                  1551          RTS                     TO COMMAND HANDLER
FEC3  30 6E               1552  STLDFT  LEAX    14,S            COMPUTE NMI COMPARE
FEC5  9F F8               1553          STX     SLEVEL          AND STORE IT
FEC7  39                  1554          RTS                     END COMMAND
FEC8                      1555  
FEC8                      1556  ******************OFFSET - COMPUTE SHORT AND LONG
FEC8                      1557  ******************                  BRANCH OFFSETS
FEC8  8D 96               1558  COFFS   BSR     CDNUM           OBTAIN INSTRUCTION ADDRESS
FECA  1F 01               1559          TFR     D,X             USE AS FROM ADDRESS
FECC  8D 92               1560          BSR     CDNUM           OBTAIN TO ADDRESS
FECE                      1561  * D=TO INSTRUCTION, X=FROM INSTRUCTION OFFSET BYTE(S)
FECE  30 01               1562          LEAX    1,X             ADJUST FOR *+2 SHORT BRANCH
FED0  34 30               1563          PSHS    Y,X             STORE WORK WORD AND VALUE ON S
FED2  A3 E4               1564          SUBD    ,S              FIND OFFSET
FED4  ED E4               1565          STD     ,S              SAVE OVER STACK
FED6  30 61               1566          LEAX    1,S             POINT FOR ONE BYTE DISPLAY
FED8  1D                  1567          SEX     SIGN            EXTEND LOW BYTE
FED9  A1 E4               1568          CMPA    ,S              ? VALID ONE BYTE OFFSET
FEDB  26 02               1569          BNE     COFNO1          BRANCH IF NOT
FEDD  3F                  1570          SWI                     SHOW ONE BYTE OFFSET
FEDE  04                  1571          FCB     OUT2HS          FUNCTION
FEDF  EE E4               1572  COFNO1  LDU     ,S              RELOAD OFFSET
FEE1  33 5F               1573          LEAU    -1,U            CONVERT TO LONG BRANCH OFFSET
FEE3  EF 84               1574          STU     ,X              STORE BACK WHERE X POINTS NOW
FEE5  3F                  1575          SWI                     SHOW TWO BYTE OFFSET
FEE6  05                  1576          FCB     OUT4HS          FUNCTION
FEE7  3F                  1577          SWI                     FORCE NEW LINE
FEE8  06                  1578          FCB     PCRLF           FUNCTION
FEE9  35 96               1579          PULS    PC,X,D          RESTORE STACK AND END COMMAND
FEEB                      1580  *H
FEEB                      1581  
FEEB                      1582  *************BREAKPOINT - DISPLAY/ENTER/DELETE/CLEAR
FEEB                      1583  *************             BREAKPOINTS
FEEB  27 23               1584  CBKPT   BEQ     CBKDSP          BRANCH DISPLAY OF JUST 'B'

FEED  17 FD F1            1585          LBSR    BLDNUM          ATTEMPT VALUE ENTRY
FEF0  27 2C               1586          BEQ     CBKADD          BRANCH TO ADD IF SO
FEF2  81 2D               1587          CMPA    #'-'            ? CORRECT DELIMITER
FEF4  26 3F               1588          BNE     CBKERR          NO, BRANCH FOR ERROR
FEF6  17 FD E8            1589          LBSR    BLDNUM          ATTEMPT DELETE VALUE
FEF9  27 03               1590          BEQ     CBKDLE          GOT ONE, GO DELETE IT
FEFB  0F FA               1591          CLR     BKPTCT          WAS 'B -', SO ZERO COUNT
FEFD  39                  1592  CBKRTS  RTS                     END COMMAND
FEFE                      1593  * DELETE THE ENTRY
FEFE  8D 40               1594  CBKDLE  BSR     CBKSET          SETUP REGISTERS AND VALUE
FF00  5A                  1595  CBKDLP  DECB                    ? ANY ENTRIES IN TABLE
FF01  2B 32               1596          BMI     CBKERR          BRANCH NO, ERROR
FF03  AC A1               1597          CMPX    ,Y++            ? IS THIS THE ENTRY
FF05  26 F9               1598          BNE     CBKDLP          NO, TRY NEXT
FF07                      1599  * FOUND, NOW MOVE OTHERS UP IN ITS PLACE
FF07  AE A1               1600  CBKDLM  LDX     ,Y++            LOAD NEXT ONE UP
FF09  AF 3C               1601          STX     -4,Y            MOVE DOWN BY ONE
FF0B  5A                  1602          DECB                    ? DONE
FF0C  2A F9               1603          BPL     CBKDLM          NO, CONTINUE MOVE
FF0E  0A FA               1604          DEC     BKPTCT          DECREMENT BREAKPOINT COUNT
FF10  8D 2E               1605  CBKDSP  BSR     CBKSET          SETUP REGISTERS AND LOAD VALUE
FF12  27 E9               1606          BEQ     CBKRTS          RETURN IF NONE TO DISPLY
FF14  30 A1               1607  CBKDSL  LEAX    ,Y++            POINT TO NEXT ENTRY
FF16  3F                  1608          SWI                     DISPLAY IN HEX
FF17  05                  1609          FCB     OUT4HS          FUNCTION
FF18  5A                  1610          DECB                    COUNT DOWN
FF19  26 F9               1611          BNE     CBKDSL          LOOP IF NGABLE RAM
FF1B  3F                  1612          SWI                     SKIP TO NEW LINK
FF1C  06                  1613          FCB     PCRLF           FUNCTIONRTS
FF1D  39                  1614          RTS
FF1E                      1615  * ADD NEW ENTRY
FF1E  8D 20               1616  CBKADD  BSR     CBKSET          SETUP REGISTERS
FF20  C1 08               1617          CMPB    #NUMBKP         ? ALREADY FULL
FF22  27 11               1618          BEQ     CBKERR          BRANCH ERROR IF SO
FF24  A6 84               1619          LDA     ,X              LOAD BYTE TO TRAP
FF26  E7 84               1620          STB     ,X              TRY TO CHANGE
FF28  E1 84               1621          CMPB    ,X              ? CHANGEABLE RAM
FF2A  26 09               1622          BNE     CBKERR          BRANCH ERROR IF NOT
FF2C  A7 84               1623          STA     ,X              RESTORE BYTE
FF2E  5A                  1624  CBKADL  DECB                    COUNT DOWN
FF2F  2B 07               1625          BMI     CBKADT          BRANCH IF DONE TO ADD IT
FF31  AC A1               1626          CMPX    ,Y++            ? ENTRY ALREADY HERE
FF33  26 F9               1627          BNE     CBKADL          LOOP IF NOT
FF35  16 FA 24            1628  CBKERR  LBRA    CMDBAD          RETURN TO ERROR PRODUCE
FF38  AF A4               1629  CBKADT  STX     ,Y              ADD THIS ENTRY
FF3A  6F 31               1630          CLR     -NUMBKP*2+1,Y   CLEAR OPTIONAL BYTE
FF3C  0C FA               1631          INC     BKPTCT          ADD ONE TO COUNT
FF3E  20 D0               1632          BRA     CBKDSP          AND NOW DISPLAY ALL OF 'EM
FF40                      1633  * SETUP REGISTERS FOR SCAN
FF40  9E 9B               1634  CBKSET  LDX     NUMBER          LOAD VALUE DESIRED
FF42  31 8D E0 6C         1635  CBKLDR  LEAY    BKPTBL,PCR      LOAD START OF TABLE
FF46  D6 FA               1636          LDB     BKPTCT          LOAD ENTRY COUNT
FF48  39                  1637          RTS                     RETURN
FF49                      1638  
FF49                      1639  *****************ENCODE  -  ENCODE A POSTBYTE
FF49  6F E2               1640  CENCDE  CLR     ,-S             DEFAULT TO NOT INDIRECT
FF4B  5F                  1641          CLRB                    ZERO POSTBYTE VALUE
FF4C  30 8C 3F            1642          LEAX    <CONV1,PCR      START TABLE SEARCH
FF4F  3F                  1643          SWI                     OBTAIN FIRST CHARACTER
FF50  00                  1644          FCB     INCHNP          FUNCTION
FF51  81 5B               1645          CMPA    #'['            ? INDIRECT HERE
FF53  26 06               1646          BNE     CEN2            BRANCH IF NOT
FF55  86 10               1647          LDA     #$10            SET INDIRECT BIT ON
FF57  A7 E4               1648          STA     ,S              SAVE FOR LATER
FF59  3F                  1649  CENGET  SWI                     OBTAIN NEXT CHARACTER
FF5A  00                  1650          FCB     INCHNP          FUNCTION
FF5B  81 0D               1651  CEN2    CMPA    #CR             ? END OF ENTRY
FF5D  27 0C               1652          BEQ     CEND1           BRANCH YES

FF5F  6D 84               1653  CENLP1  TST     ,X              ? END OF TABLE
FF61  2B D2               1654          BMI     CBKERR          BRANCH ERROR IF SO
FF63  A1 81               1655          CMPA    ,X++            ? THIS THE CHARACTER
FF65  26 F8               1656          BNE     CENLP1          BRANCH IF NOT
FF67  EB 1F               1657          ADDB    -1,X            ADD THIS VALUE
FF69  20 EE               1658          BRA     CENGET          GET NEXT INPUT
FF6B  30 8C 49            1659  CEND1   LEAX    <CONV2,PCR      POINT AT TABLE 2
FF6E  1F 98               1660          TFR     B,A             SAVE COPY IN A
FF70  84 60               1661          ANDA    #$60            ISOLATE REGISTER MASK
FF72  AA E4               1662          ORA     ,S              ADD IN INDIRECTION BIT
FF74  A7 E4               1663          STA     ,S              SAVE BACK AS POSTBYTE SKELETON
FF76  C4 9F               1664          ANDB    #$9F            CLEAR REGISTER BITS
FF78  6D 84               1665  CENLP2  TST     ,X              ? END OF TABLE
FF7A  27 B9               1666          BEQ     CBKERR          BRANCH ERROR IF SO
FF7C  E1 81               1667          CMPB    ,X++            ? SAME VALUE
FF7E  26 F8               1668          BNE     CENLP2          LOOP IF NOT
FF80  E6 1F               1669          LDB     -1,X            LOAD RESULT VALUE
FF82  EA E4               1670          ORB     ,S              ADD TO BASE SKELETON
FF84  E7 E4               1671          STB     ,S              SAVE POSTBYTE ON STACK
FF86  30 E4               1672          LEAX    ,S              POINT TO IT
FF88  3F                  1673          SWI                     SEND OUT AS HEX
FF89  04                  1674          FCB     OUT2HS          FUNCTION
FF8A  3F                  1675          SWI                     TO NEXT LINE
FF8B  06                  1676          FCB     PCRLF           FUNCTION
FF8C  35 84               1677          PULS    PC,B            END OF COMMAND
FF8E                      1678  
FF8E                      1679  * TABLE ONE DEFINES VALID INPUT IN SEQUENCE
FF8E  41 04 42 05 44 06 + 1680  CONV1   FCB     'A',$04,'B',$05,'D',$06,'H',$01
FF96  48 01 48 01 48 00 + 1681          FCB     'H',$01,'H',$01,'H',$00,',',$00
FF9E  2D 09 2D 01 53 70 + 1682          FCB     '-',$09,'-',$01,'S',$70,'Y',$30
FFA6  55 50 58 10 2B 07 + 1683          FCB     'U',$50,'X',$10,'+',$07,'+',$01
FFAE  50 80 43 00 52 00 + 1684          FCB     'P',$80,'C',$00,'R',$00,']',$00
FFB6  FF                  1685          FCB     $FF             END OF TABLE
FFB7                      1686  *CONV2 USES ABOVE CONVERSION TO SET POSTBYTE
FFB7                      1687  *                             BIT SKELETON.
FFB7  10 84 11 00         1688  CONV2   FDB     $1084,$1100     R,        H,R
FFBB  12 88 13 89         1689          FDB     $1288,$1389     HH,R      HHHH,R
FFBF  14 86 15 85         1690          FDB     $1486,$1585     A,R       B,R
FFC3  16 8B 17 80         1691          FDB     $168B,$1780     D,R       ,R+
FFC7  18 81 19 82         1692          FDB     $1881,$1982     ,R++      ,-R
FFCB  1A 83 82 8C         1693          FDB     $1A83,$828C     ,--R      HH,PCR
FFCF  83 8D 03 9F         1694          FDB     $838D,$039F     HHHH,PCR  [HHHH]
FFD3  00                  1695          FCB     0               END OF TABLE
FFD4                      1696  
FFD4                      1697  ****************************************************
FFD4                      1698  *            DEFAULT INTERRUPT TRANSFERS           *
FFD4                      1699  ****************************************************
FFD4  6E 9D DF EE         1700  RSRVD   JMP     [VECTAB+.RSVD,PCR]      RESERVED VECTOR
FFD8  6E 9D DF EC         1701  SWI3    JMP     [VECTAB+.SWI3,PCR]      SWI3 VECTOR
FFDC  6E 9D DF EA         1702  SWI2    JMP     [VECTAB+.SWI2,PCR]      SWI2 VECTOR
FFE0  6E 9D DF E8         1703  FIRQ    JMP     [VECTAB+.FIRQ,PCR]      FIRQ VECTOR
FFE4  6E 9D DF E6         1704  IRQ     JMP     [VECTAB+.IRQ,PCR]       IRQ VECTOR
FFE8  6E 9D DF E4         1705  SWI     JMP     [VECTAB+.SWI,PCR]       SWI VECTOR
FFEC  6E 9D DF E2         1706  NMI     JMP     [VECTAB+.NMI,PCR]       NMI VECTOR
FFF0                      1707  
FFF0                      1708  ******************************************************
FFF0                      1709  *             ASSIST09 HARDWARE VECTOR TABLE
FFF0                      1710  *  THIS TABLE IS USED IF THE ASSIST09 ROM ADDRESSES
FFF0                      1711  *  THE MC6809 HARDWARE VECTORS.
FFF0                      1712  ******************************************************
FFF0                      1713          ORG     ROMBEG+ROMSIZ-16        SETUP HARDWARE VECTORS
FFF0  FF D4               1714          FDB     RSRVD           RESERVED SLOT
FFF2  FF D8               1715          FDB     SWI3            SOFTWARE INTERRUPT 3
FFF4  FF DC               1716          FDB     SWI2            SOFTWARE INTERRUPT 2
FFF6  FF E0               1717          FDB     FIRQ            FAST INTERRUPT REQUEST
FFF8  FF E4               1718          FDB     IRQ             INTERRUPT REQUEST
FFFA  FF E8               1719          FDB     SWI             SOFTWARE INTERRUPT
FFFC  FF EC               1720          FDB     NMI             NON-MASKABLE INTERRUPT

FFFE  F8 37               1721          FDB     RESET           RESTART
0000                      1722  

SYMBOL TABLE:

.ACIA   -002E   .AVTBL  -0000   .BSDTA  -0024   .BSOFF  -0026   .BSON   -0022

.CIDTA  -0016   .CIOFF  -0018   .CION   -0014   .CMDL1  -0002   .CMDL2  -002C

.CODTA  -001C   .COOFF  -001E   .COON   -001A   .ECHO   -0032   .EXPAN  -002A

.FIRQ   -000A   .HSDTA  -0020   .IRQ    -000C   .NMI    -0010   .PAD    -0030

.PAUSE  -0028   .PTM    -0034   .RESET  -0012   .RSVD   -0004   .SWI    -000E

.SWI2   -0008   .SWI3   -0006   ACIA    -E008   ADDR    -DF9E   ARMBK2  -FDA7
ARMBLP  -FD8E   ARMLOP  -FDAC   ARMNSW  -FD9D   BASEPG  -DF9D   BELL    -0007
BKPTBL  -DFB2   BKPTCT  -DFFA   BKPTOP  -DFA2   BLD2    -F815   BLD3    -F821
BLDBAD  -FD46   BLDHEX  -FD4D   BLDHXC  -FD4F   BLDHXI  -FD49   BLDNNB  -FCDF
BLDNUM  -FCE1   BLDRTN  -F835   BLDSHF  -FD58   BLDVTR  -F800   BRKPT   -000A
BSDCMP  -FB6A   BSDEOL  -FB70   BSDLD1  -FB40   BSDLD2  -FB42   BSDNXT  -FB60
BSDPUN  -FB92   BSDSRT  -FB6E   BSDTA   -FB38   BSOFF   -FB27   BSOFLP  -FB33
BSON    -FB1B   BSON2   -FB22   BSPEOF  -FBEF   BSPGO   -FBA3   BSPMRE  -FBC6
BSPOK   -FBAF   BSPSTR  -FBEC   BSPUN2  -FBE7   BSPUNC  -FBE9   BYTE    -FB75
BYTHEX  -FB89   BYTRTS  -FB88   CAN     -0018   CBKADD  -FF1E   CBKADL  -FF2E
CBKADT  -FF38   CBKDLE  -FEFE   CBKDLM  -FF07   CBKDLP  -FF00   CBKDSL  -FF14
CBKDSP  -FF10   CBKERR  -FF35   CBKLDR  -FF42   CBKPT   -FEEB   CBKRTS  -FEFD
CBKSET  -FF40   CCALBS  -FE7B   CCALL   -FDB9   CDBADN  -FE6E   CDCNT   -FE5A
CDISP   -FE43   CDISPS  -FE51   CDNUM   -FE60   CDOT    -FEA8   CEN2    -FF5B
CENCDE  -FF49   CEND1   -FF6B   CENGET  -FF59   CENLP1  -FF5F   CENLP2  -FF78
CGO     -FD80   CGOBRK  -FDBF   CHKABT  -FA58   CHKRTN  -FA61   CHKSEC  -FA60
CHKWT   -FA62   CIDTA   -FADC   CIOFF   -FAF0   CION    -FAE6   CIRTN   -FAE5
CLOAD   -FE8F   CLVDFT  -FE9B   CLVOFS  -FE92   CMD     -F8F7   CMD2    -F935
CMD3    -F948   CMDBAD  -F95C   CMDCMP  -F977   CMDDDL  -F901   CMDFLS  -F96C
CMDGOT  -F94D   CMDMEM  -F990   CMDNEP  -F8F9   CMDNOL  -F90A   CMDSCH  -F953
CMDSIZ  -F96F   CMDSME  -F967   CMDTB2  -F99B   CMDTBL  -F99C   CMDXQT  -F987
CMEM    -FDC3   CMEM2   -FDC8   CMEM4   -FDD1   CMEMN   -FDC6   CMENUM  -FDE0
CMESTR  -FDEC   CMNOTB  -FE02   CMNOTC  -FDE8   CMNOTL  -FE0E   CMNOTQ  -FDF8
CMNOTU  -FE1C   CMPADP  -FE18   CMPADS  -FE16   CMSPCE  -FDFE   CNULLS  -FEB7
CNVGOT  -FD74   CNVHEX  -FD62   CNVOK   -FD76   CNVRTS  -FD78   CODTA   -FAF1
CODTAD  -FB0F   CODTAO  -FB12   CODTLP  -FB07   CODTPD  -FB03   CODTRT  -FB0D
COFFS   -FEC8   COFNO1  -FEDF   CONV1   -FF8E   CONV2   -FFB7   COOFF   -FAF0
COON    -FAE6   CPUNCH  -FE71   CR      -000D   CREG    -FC4A   CSTLEV  -FEBC
CTRACE  -FEA4   CTRCE3  -FEAA   CVER    -FEA1   CWINDO  -FE3E   DELIM   -DF8E
DFTCHP  -0000   DFTNLP  -0005   DLE     -0010   EOT     -0004   ERRMSG  -FABD
ERROR   -FACE   EXP1    -FCE9   EXP2    -FD07   EXPADD  -FD23   EXPCDL  -FD17
EXPCHM  -FD2B   EXPDLM  -FCEB   EXPRTN  -FD05   EXPSUB  -FD36   EXPTDI  -FD0D
EXPTDL  -FD0F   EXPTRM  -FD42   FIRQ    -FFE0   FIRQR   -FABC   GOADDR  -FD83
GONDFT  -FDA2   HIVTR   -0034   HSBLNK  -FC00   HSDRTN  -FC47   HSDTA   -FBFC
HSHCHR  -FC2B   HSHCOK  -FC35   HSHDOT  -FC33   HSHLNE  -FC14   HSHNXT  -FC20
HSHTTL  -FC06   INCHNP  -0000   INITVT  -F844   INTVE   -F87D   INTVS   -F870
IRQ     -FFE4   IRQR    -FAD8   LASTOP  -DF99   LDDP    -FAC1   LF      -000A
MISFLG  -DF8F   MONITR  -0008   MSHOWP  -FA79   MUPBAD  -FE36   MUPDAT  -FE2B
NMI     -FFEC   NMICON  -FAB7   NMIR    -FA7D   NMITRC  -FAB0   NUMBER  -DF9B
NUMBKP  -0008   NUMFUN  -000B   NUMVTR  -001B   OUT2HS  -0004   OUT4HS  -0005
OUTCH   -0001   PAUSE   -000B   PAUSER  -DFFC   PCNTER  -DF93   PCRLF   -0006
PDATA   -0003   PDATA1  -0002   PROMPT  -003E   PRTADR  -FE21   PSTACK  -DF95
PTM     -E000   PTMC13  -E000   PTMC2   -E001   PTMSTA  -E001   PTMTM1  -E002
PTMTM2  -E004   PTMTM3  -E006   RAMOFS  -E700   READ    -FD79   REG4    -FC94
REGAGN  -FCC3   REGCHG  -FC70   REGCNG  -FC9D   REGMSK  -FC50   REGNXC  -FCB1
REGP1   -FC78   REGP2   -FC81   REGP3   -FC92   REGPRS  -FAB3   REGPRT  -FC6F
REGRTN  -FC9B   REGSKP  -FCAA   REGTF1  -FCC9   REGTF2  -FCD6   REGTWO  -FCBB
RESET   -F837   RESET2  -F83D   ROM2OF  -F000   ROM2WK  -DF66   ROMBEG  -F800
ROMSIZ  -0800   RSRVD   -FFD4   RSRVDR  -FAD8   RSTACK  -DF97   RTI     -FABC
RTS     -FAF0   SEND    -F9EC   SIGNON  -F8C9   SKIP2   -008C   SLEVEL  -DFF8
SPACE   -0007   STACK   -DF51   STLDFT  -FEC3   SWI     -FFE8   SWI2    -FFDC
SWI2R   -FAD8   SWI3    -FFD8   SWI3R   -FAD8   SWIBFL  -DFFB   SWICNT  -DF90
SWIDNE  -F8B5   SWILP   -F8A8   SWIR    -F895   SWIVTB  -F87D   TRACEC  -DF91
TSTACK  -DF51   VCTRSW  -0009   VECTAB  -DFC2   WINDOW  -DFA0   WORKPG  -DF00
XQCIDT  -FA72   XQPAUS  -FA6E   ZBKCMD  -FAD5   ZBKPNT  -FAD3   ZIN2    -FA2A
ZINCH   -FA11   ZINCHP  -FA0F   ZMONT2  -F8E6   ZMONTR  -F8D2   ZOT2HS  -F9F2
ZOT4HS  -F9F0   ZOTCH1  -FA2E   ZOTCH2  -FA37   ZOTCH3  -FA39   ZOUT2H  -F9D9
ZOUTHX  -F9E6   ZPAUSE  -FA4E   ZPCRLF  -FA3D   ZPCRLS  -FA3C   ZPDATA  -FA40
ZPDTA1  -FA48   ZPDTLP  -FA46   ZSPACE  -F9F6   ZVSWTH  -F9FA



  Symbol   Define       References
---------- ------ ----------------------
.ACIA          92 
.AVTBL         69 
.BSDTA         87 
.BSOFF         88 
.BSON          86 
.CIDTA         80 
.CIOFF         81 
.CION          79 
.CMDL1         70 
.CMDL2         91 
.CODTA         83 
.COOFF         84 
.COON          82 
.ECHO          94 
.EXPAN         90 
.FIRQ          74 
.HSDTA         85 
.IRQ           75 
.NMI           77 
.PAD           93 
.PAUSE         89 
.PTM           95 
.RESET         78 
.RSVD          71 
.SWI           76 
.SWI2          73 
.SWI3          72 
ACIA           21   253   822   834   850 
ADDR          130  1236  1388  1389  1399  1418  1428  1434  1445  1453  1457 
ARMBK2       1366   770  1354 
ARMBLP       1353  1357 
ARMLOP       1368  1374 
ARMNSW       1361  1359 
BASEPG        132   183   781 
BELL           33   779 
BKPTBL        124  1635 
BKPTCT        118   383  1367  1591  1604  1631  1636 
BKPTOP        126 
BLD2          189   193 
BLD3          195   198 
BLDBAD       1285  1336 
BLDHEX       1298  1247 
BLDHXC       1299   418 
BLDHXI       1296  1230 
BLDNNB       1216  1161  1394 
BLDNUM       1219  1283  1489  1585  1589 
BLDRTN        204   202 
BLDSHF       1304  1308 
BLDVTR        180   215 
BRKPT          63  1381 
BSDCMP        941   939 
BSDEOL        945   937 
BSDLD1        916   919   946 
BSDLD2        918   925 
BSDNXT        936   942 
BSDPUN        974   910 
BSDSRT        943   923   947 
BSDTA         908   247  1505 
BSOFF         888   248  1507 
BSOFLP        896   897 
BSON          877   246  1504 
BSON2         881   879 
BSPEOF       1030  1018 
BSPGO         984  1017 
BSPMRE       1006  1008 
BSPOK         989   987 
BSPSTR       1029   994 
BSPUN2       1026  1000  1002  1003  1006 
BSPUNC       1027  1014 
BYTE          950   927   930   932   936 
BYTHEX        962   950   953 
BYTRTS        960   965 
CAN            37   708   715  1335 
CBKADD       1616  1586 
CBKADL       1624  1627 
CBKADT       1629  1625 
CBKDLE       1594  1590 
CBKDLM       1600  1603 
CBKDLP       1595  1598 
CBKDSL       1607  1611 
CBKDSP       1605  1584  1632 
CBKERR       1628  1588  1596  1618  1622  1654  1666 



CBKLDR       1635   300   380  1351  1366 
CBKPT        1584   500 
CBKRTS       1592  1606 
CBKSET       1634  1594  1605  1616 
CCALBS       1504  1524 
CCALL        1377   503 
CDBADN       1496  1490  1492  1509 
CDCNT        1482  1480 
CDISP        1471   506 
CDISPS       1478  1475 
CDNUM        1489  1364  1387  1466  1471  1476  1499  1501  1519  1532  1543 
                   1549  1558  1560 
CDOT         1534   405  1362 
CEN2         1651  1646 
CENCDE       1640   509 
CEND1        1659  1652 
CENGET       1649  1658 
CENLP1       1653  1656 
CENLP2       1665  1668 
CGO          1341   512 
CGOBRK       1380  1382 
CHKABT        706   698   761 
CHKRTN        711   707 
CHKSEC        710   716 
CHKWT         712   709   714 
CIDTA         822   240   722 
CIOFF         841   241 
CION          832   239   345 
CIRTN         827   825 
CLOAD        1513   515 
CLVDFT       1521  1518 
CLVOFS       1516  1513  1527 
CMD           377   351   436 
CMD2          412   422 
CMD3          421   419 
CMDBAD        432   461  1285  1496  1628 
CMDCMP        447   452 
CMDDDL        384   388 
CMDFLS        441   450 
CMDGOT        424   413 
CMDMEM        460   417 
CMDNEP        380   797 
CMDNOL        389   381   385   459 
CMDSCH        427   431   442 
CMDSIZ        443   440 
CMDSME        438   428 
CMDTB2        493   251 
CMDTBL        497   230 
CMDXQT        456   407   410   464 
CMEM         1387   518 
CMEM2        1389  1421  1438 
CMEM4        1394  1401  1405 
CMEMN        1388   462 
CMENUM       1402  1395 
CMESTR       1409  1414 
CMNOTB       1423  1417 
CMNOTC       1407  1398 
CMNOTL       1431  1424 
CMNOTQ       1416  1408 
CMNOTU       1440  1432 
CMPADP       1437   408   462  1429 
CMPADS       1435  1441 
CMSPCE       1419  1411 
CNULLS       1543   521 
CNVGOT       1328  1322 
CNVHEX       1319   964  1299 
CNVOK        1329  1309 
CNVRTS       1330  1284  1300  1320  1324  1326  1369 
CODTA         849   243   565 
CODTAD        866   869 
CODTAO        867   851   861 
CODTLP        861   863 
CODTPD        858   856 
CODTRT        864   853 
COFFS        1558   524 
COFNO1       1572  1569 
CONV1        1680  1642 
CONV2        1688  1659 
COOFF         842   244 
COON          833   242   346 
CPUNCH       1499   527 
CR             35   424   618   664   855  1031  1163  1182  1425  1493  1651 
CREG         1099   530 
CSTLEV       1548   533 



CTRACE       1532   536 
CTRCE3       1535   763 
CVER         1527   539 
CWINDO       1466   542 
DELIM         150   748   754  1220  1233  1253 
DFTCHP         23   254 
DFTNLP         24   254 
DLE            36   852 
EOT            32   340   649   681   735   779  1029  1031 
ERRMSG        779   433   786 
ERROR         786   311 
EXP1         1229   250 
EXP2         1247  1231  1248 
EXPADD       1263  1279 
EXPCDL       1257  1249  1266 
EXPCHM       1267  1259 
EXPDLM       1230  1234 
EXPRTN       1245  1254  1272 
EXPSUB       1273  1268 
EXPTDI       1251  1270 
EXPTDL       1252  1238  1241  1244 
EXPTRM       1283  1260  1273 
FIRQ         1703  1703  1717 
FIRQR         813   234 
GOADDR       1346  1341  1377 
GONDFT       1364  1348 
HIVTR          97   589 
HSBLNK       1043  1046 
HSDRTN       1089  1059  1083 
HSDTA        1040   245  1088  1482 
HSHCHR       1073  1081 
HSHCOK       1078  1076 
HSHDOT       1077  1074 
HSHLNE       1057  1087 
HSHNXT       1065  1068 
HSHTTL       1048  1056 
INCHNP         53   917   921   963  1334  1644  1650 
INITVT        230   185 
INTVE         261   194 
INTVS         253   194 
IRQ          1704  1704  1718 
IRQR          805   235 
LASTOP        136   749  1536 
LDDP          781   294   737   806 
LF             34   620   635   666  1031  1423 
MISFLG        148   399   616   738   769   883   894  1361 
MONITR         61   219 
MSHOWP        735   745 
MUPBAD       1459  1456 
MUPDAT       1453  1403  1413 
NMI          1706  1706  1720 
NMICON        769   739 
NMIR          737   237 
NMITRC        763   741   744 
NUMBER        134   398   463  1176  1252  1257  1262  1264  1275  1296  1297 
                   1305  1306  1402  1494  1634 
NUMBKP         26   123   125   386  1355  1371  1617  1630 
NUMFUN         65   310 
NUMVTR         96   121   187 
OUT2HS         57  1066  1153  1571  1674 
OUT4HS         58   751  1062  1150  1449  1576  1609 
OUTCH          54   393   882   890   893   980  1079  1139  1143  1393  1427 
                   1462 
PAUSE          64   721 
PAUSER        114   249 
PCNTER        142   390  1239 
PCRLF          59   378  1041  1058  1090  1158  1436  1578  1613  1676 
PDATA          56   349   788   996  1020 
PDATA1         55   435   747 
PROMPT         25   391 
PRTADR       1445  1437 
PSTACK        140   395   432 
PTM            22    39    40    41    42    43    44   256   350   352   353 
                    355   356   358  1537  1539 
PTMC13         40   356 
PTMC2          41   355   358 
PTMSTA         39 
PTMTM1         42   352   353  1539 
PTMTM2         43 
PTMTM3         44 
RAMOFS         18   108 
READ         1333   404   421  1255  1298  1409 
REG4         1154  1173  1183 
REGAGN       1186  1164 



REGCHG       1132  1101 
REGCNG       1161  1146 
REGMSK       1120  1134 
REGNXC       1174  1162 
REGP1        1135  1140  1156 
REGP2        1141  1137 
REGP3        1152  1148 
REGPRS        765   752   796 
REGPRT       1131   765  1099 
REGRTN       1159  1198 
REGSKP       1169  1172 
REGTF1       1188  1191 
REGTF2       1194  1197 
REGTWO       1180  1178 
RESET         214   238  1721 
RESET2        216   220 
ROM2OF         20   199 
ROM2WK        152 
ROMBEG         17    20   108   164  1713 
ROMSIZ         19    20  1713 
RSRVD        1700  1714 
RSRVDR        806   231 
RSTACK        138   342   785 
RTI           771   813 
RTS           838   784   841   842 
SEND          565   621   637   665   679 
SIGNON        339   347 
SKIP2          46   860  1151  1217  1520 
SLEVEL        120   743  1550  1553 
SPACE          60  1044  1051  1053  1070  1170  1420 
STACK         155   214 
STLDFT       1552  1548 
SWI          1705  1705  1719 
SWI2         1702  1702  1716 
SWI2R         803   233 
SWI3         1701  1701  1715 
SWI3R         804   232 
SWIBFL        116   298   308  1360 
SWICNT        146   293   638   740 
SWIDNE        308   299   303 
SWILP         302   305 
SWIR          293   236 
SWIVTB        280   280   281   282   283   284   285   286   287   288   289 
                    290   291   314 
TRACEC        144   400   756   759  1533 
TSTACK        154  1186 
VCTRSW         62 
VECTAB        122   180   345   346   350   426   429   565   591   622   721 
                    722   822   834   850   854   857   974   978   982  1022 
                   1221  1482  1504  1505  1507  1537  1544  1700  1701  1702 
                   1703  1704  1705  1706 
WINDOW        128  1242  1467 
WORKPG        108   109   110 
XQCIDT        722   609   706   713 
XQPAUS        721   608   697   712   866 
ZBKCMD        797   753   755   757   760   762 
ZBKPNT        796   290   307   807 
ZIN2          622   619 
ZINCH         609   280   612   614 
ZINCHP        608   610 
ZMONT2        350   344 
ZMONTR        342   288 
ZOT2HS        568   284 
ZOT4HS        567   285 
ZOTCH1        633   281 
ZOTCH2        637   579 
ZOTCH3        638   590   595   597   617   623   701 
ZOUT2H        554   567   568  1027  1390 
ZOUTHX        561   558  1049 
ZPAUSE        697   291 
ZPCRLF        651   286 
ZPCRLS        649   634   651 
ZPDATA        664   283 
ZPDTA1        680   282 
ZPDTLP        679   636   682 
ZSPACE        578   287 
ZVSWTH        588   289 

Symbols: 319, References: 610


































